

HISTORIC.BAS	0.66	4/12/2025
--------------	------	-----------

DESCRIPTION :

- Program for Generating Historical Time Tables

REFERENCES :

9. Remember the former things of old: for I [am] God, and [there is] none else; [I am] God, and [there is] none like me,
 10. Declaring the end from the beginning, and from ancient times [the things] that are not [yet] done, saying, My counsel shall stand, and I will do all my pleasure:
 (Isaiah 46)
 18. For verily I say unto you, Till heaven and earth pass, one jot or one tittle shall in no wise pass from the law, till all be fulfilled.
 (St. Matthew 5)

The Bible

"The Bible, Authorized Version" by King James 1769, and Webster Update 1833, Oxford University Press, 1994

[Zem1987]

Zemanek H.: "Kalender und Chronologie", (Calendar and Chronology)
 R. Oldenbourg Verlag Munich, Vienna, 4th improved edition, (1987)

ACKNOWLEDGEMENT :

- The following persons have contributed to the success of this work:
- Gebhard Böhm, parson and scripture teacher: Source analysis 1986
 - Joachim Wohlfahrt, teacher at Schubart second. school Aalen, 1986-1989
 - Eckhard Walter, Stetten, Sinsheim, and Adelshofen, since 1989
 - Markus Britsch, Pforzheim: Mountain camp 1992 with first calc. results
 - Herbert Henheik, Tübingen: Discussion and literature, 1994
 - Stefan Müller, Böblingen: Computer camp 1994 in Lausnitz
 - Karl-Heinz Eismann, Lausnitz: Camp 1994, and meeting 2000
 - Georg Grittmann, Sinsheim: Publication via computer mission CD 2000
 - Andreas Mulack, Langenorla: Meeting 2000 with algorithmic presentation
 - Gerd Baumann, Herrsching: Discussion via astronomy package, since 2001
 - Claus-Edwin Jost, Schw. Gmünd: Check of the file HISTORIC.BAT, 2005
 - Thomas Antoni, Erlangen: English version of QBASIC, 2007
 - Wolfgang Bay, Aalen: Help to find the relevant English Bibles, 2009
 - Wilhelm Hecke, Ellwangen: Representation of English Bible Cites, 2009
 - Johannes Stimmer, Aalen: Getting the DOS Box 0.73 for Windows, 2009
 - Camil Pogolski, Aalen: Getting the DOS Box for Linux and Mac, 2009
 - Heinz Döbele, Aalen: Chronological summary of the Holy Bible, 2016
 - Utho Maier, Aalen: Adjustment of the current Portable DOS Box, 2025

HANDLING :

11/29/1994 - 12/19/1994	Norbert Südland, Aalen
8/30/1999 - 2/27/2000	Norbert Südland, Ulm/Donau
7/15/2001 - 9/15/2001	Norbert Südland, Munich
8/31/2002 - 12/19/2007	Norbert Südland, Aalen
1/20/2009 - 8/13/2009	Norbert Südland, Aalen

8/18/2012 - 3/22/2017	Norbert Südland, Aalen
10/21/2022 - 4/12/2025	Norbert Südland, Aalen

.....

PREPARATION

First Commands:

```
OPTION BASE 1           'Arrays start by index No.`1`!
'OPTION EXPLICIT        'This is useful with Visaul Basic
COMMON HistoricChoice%, WorkingPlace$, WorkingTime$, CountingMode%
```

Constants:

```
CONST VersionProgrammingDate$ = "Version 0.66 from 04/12/2025"
CONST ConfigurationFile$ = "HISTORIC.CFG"
```

Announce Subroutines:

Basic Menu:

```
DECLARE SUB Pause ()
DECLARE SUB StartingPage (Answer%, InputSuccess%)

DECLARE FUNCTION Dating% (HistoricChoice%, Choice%, Term%)
DECLARE FUNCTION KeyInput$ ()
DECLARE FUNCTION Question$ (Text$, Default$, InputSuccess%)
DECLARE FUNCTION SetUp% (Data$, Position%, HistoricChoice%, InputSuccess%)
DECLARE FUNCTION SingleInput% (Data$, c%, Colors%, InputSuccess%)
DECLARE FUNCTION StartingText$ (Answer%)
```

Change Data:

```
DECLARE SUB Add (Summand1$, Kind$, Direction$, Def$, ONum1%, ONum2%, Nxt$)
DECLARE SUB Change (Data$, Variable$, Contents$)
DECLARE SUB NewStatus (DateValue$)
DECLARE SUB WriteTo (File%, DataSetLength%, Position%, Data$)

DECLARE FUNCTION Editor$ (Default$, Length%, Mode%, InputSuccess%)
DECLARE FUNCTION FillUpText$ (Text$, WantedLength%, Character$)
DECLARE FUNCTION SimultaneousnessCheck$ (Text$)
```

```

DECLARE FUNCTION ToFile$ (symbol$)
DECLARE FUNCTION ToUser$ (symbol$)

```

```

'-----'
'|                                     Represent Data:                                     |'
'-----'

```

```

DECLARE SUB LINEINPUT (File%, DataLine$)
DECLARE SUB Present (Colors%, Quest$, Buffer$, Offs%, k$, Area%, Of%, ly%)
DECLARE SUB PresentDate (Data$, c%, Colors%, HistoricChoice%)
DECLARE SUB PresentLine (Data$, c%, Colors%)

```

```

DECLARE FUNCTION CurrentDate$ ()
DECLARE FUNCTION NameEntry$ (Position&)
DECLARE FUNCTION Overview& (HistoricChoice%, Default&, InputSuccess%)
DECLARE FUNCTION Moment% (symbol$)
DECLARE FUNCTION Load$ (File%, DataSetLength%, Position&)
DECLARE FUNCTION Number% (Text$, start%, finish%)
DECLARE FUNCTION OrdinaryNumber$ (Num%)
DECLARE FUNCTION SIZEOF% (Text$)
DECLARE FUNCTION STRLEN% (Text$, EndCharacter$)
DECLARE FUNCTION Part$ (Data$, Variable$)
DECLARE FUNCTION WeekDay% (Ctry%, y%, m%, d%)

```

```

'-----'
'|                                     Conversions:                                     |'
'-----'

```

```

DECLARE SUB RestSystem (Sign%, Year%, Month%, Day%)
DECLARE SUB TermToNumber (Term$, Sign%, Year%, Month%, Day%)
DECLARE SUB Reverse (s1%, y1%, m1%, d1%, s2%, y2%, m2%, d2%, Direction$)
DECLARE SUB TimeShift (s1%, y1%, m1%, d1%, s$, s2%, y2%, m2%, d2%, d$)

```

```

DECLARE FUNCTION NumberToTerm$ (Sign%, Year%, Month%, Day%)

```

```

DEF FNMAX (a, b) = (a + b) / 2 + ABS(a - b) / 2
DEF FNMIN (a, b) = (a + b) / 2 - ABS(a - b) / 2

```

```

'|                                     Global Variables:                                     |'
'|-----'

```

```

'Set Stack Size:
'-----'

```

```

IF CountingMode% = 0 THEN
    CLEAR , , 4096
END IF

```

```

'DIM AS STRING:
'-----'

```

```

DIM SHARED EmptyCharacter$
DIM SHARED GlobalDate$

```

```

'DIM AS INTEGER:
'-----'

```

```
DIM SHARED GlobWeekDay%
DIM SHARED GlobVarNumber%      'Number of variables for struct description
DIM SHARED InputFile%          'Reference number of the input file
DIM SHARED DataLength%         'Data length of an input line
```

'DIM AS LONG:

! _ _ _ _ _ !

```
DIM SHARED GIL&          'Line number of the input file
```

'Dynamic Data Arrays:

! ----- !

```
REDIM SHARED GVBEGIN%(1)      'Variable begin for MID$ access
```

```
REDIM SHARED GVLength%(1)      'Variable length for MID$ access
```

```
REDIM SHARED GVName$(1)      'Variable name    for Part$ access
```

```
REDIM SHARED GVType$(1)      'Variable type    for Part$ access
```

Local Variables:

```
'-----'
'|          Preparation:          |
'|-----'
```

'DIM AS STRING:

! _____ !

DIM Century\$

```
DIM InputFileName$
```

```
DIM HistoricChoice$
```

DIM m\$ 'month

DIM d\$ ' day

DIM y\$ 'year

DIM Year\$

```
DIM Text$ 'data buffer
```

DIM result\$

```
DIM DateValue$
```

DIM Data\$

DIM SwapTerm\$

DIM Comparison\$

```
DIM Buffer$
```

DIM DataLine\$

DIM Variable\$

```
DIM Summand1$
```

DIM Summand2\$

DIM Difference\$

DIM Term\$

DIM Status\$

DIM Simultaneous\$

```
DIM CopyOfFile$
```

DIM PositionFile\$

DIM Answer\$

DIM Default\$

'DIM AS INTEGER:

! ----- !

DIM Configuration%

DIM ReadingTrial%

```

DIM Position%
DIM Length%
DIM Ctry%           'century
DIM y%             'year
DIM Year%
DIM m%             'month
DIM d%             'day
DIM x%             'column position
DIM ly%            'line position
DIM InputSuccess%
DIM Success%
DIM StopFlag%
DIM OK%
DIM New%
DIM Changed%
DIM Different%
DIM Escape%
DIM Copy%
DIM Offset%
DIM DirFile%
DIM HQLFile%        'HQL = Historical Source (German abbreviation)

```

```
'DIM AS LONG INTEGER:
```

```
'-----'
```

```

DIM c&              'counter
DIM Choice&
DIM Template&
DIM Target&
DIM Quantity&

```

```

' |||-----|||
' |||                MAIN PART                |||
' |||-----|||

```

```
ON ERROR GOTO ErrorHandler
```

```
'Menu to Choose:
```

```
'-----'
```

```
GOSUB C0           'Preparation
```

```
DO
```

```
    InputSuccess% = 1
```

```
    StartingPage HistoricChoice%, InputSuccess%
```

```
    IF InputSuccess% = 1 THEN
```

```
        ON HistoricChoice% GOSUB C1, C2, C3, C4, C5, C6, C7, C8, C9
```

```
    END IF
```

```
    LOOP UNTIL HistoricChoice% = 0    'Explicit program termination is needed!
```

```
'====='
```

```
ProgramEnd:
```

```
'====='
```

```
    GOSUB SaveConfiguration
```

```
    SYSTEM
```

```
'----- END OF THE MAIN PART -----'
```

```

' |||-----|||
' |||                ERROR HANDLING                |||
' |||-----|||

```

```
'===== '
ErrorHandling:
'===== '
SELECT CASE ERR
CASE 5          'Illegal function call
  SELECT CASE ERL
  CASE 12, 13, 14, 15
    RESUME 10
  CASE 107
    RESUME 109
  CASE ELSE
    Pause
    RESUME NEXT
  END SELECT
CASE 7          'Out of memory
  SELECT CASE ERL
  CASE 5          'Set environmental variable
    RESUME NEXT
  CASE 11
    Text$ = "Please fix the current century via the DOS command "
    Text$ = Text$ + CHR$(34) + "SET CENTURY=" + Century$ + CHR$(34) + "."
    Present 1, Text$, "", 0, "C", 1, 1, 9
    Pause
    RESUME NEXT
  CASE ELSE
    Pause
    RESUME NEXT
  END SELECT
CASE 9          'Subscript out of range
  SELECT CASE ERL
  CASE 801        'Function `SIZEOF%()`
    PRINT "in the position file "; CHR$(34); PositionFile$; CHR$(34); "."
    PRINT
    COLOR 0, 7
    PRINT " The program will be terminated now! ";
    COLOR 7, 0
    Pause
    GOTO ProgramEnd
  CASE ELSE
    Pause
    RESUME NEXT
  END SELECT
CASE 52          'Bad file name or number
  SELECT CASE ERL
  CASE 3, 4        'Read in the configuration file
    RESUME NEXT
  CASE ELSE
    Pause
    RESUME NEXT
  END SELECT
CASE 53          'File not found
  SELECT CASE ERL
  CASE 2          'Open the configuration file
    PRINT "The configuration file "; CHR$(34); ConfigurationFile$; CHR$(34)
    PRINT " is missing and will be generated."
    RESUME NEXT
```

```
CASE 6
  PRINT "The structure file "; CHR$(34); "HISTORIC.STR"; CHR$(34); " is";
  PRINT " missing."
  PRINT
  COLOR 0, 7
  PRINT " Please start the program via HISTORIC.BAT! "
  COLOR 7, 0
  Pause
  GOTO ProgramEnd
CASE 7      'prepare position file
  PRINT "The program part "; CHR$(34); "STRUCT.BAS"; CHR$(34); " is";
  PRINT " missing. "
  PRINT
  COLOR 0, 7
  PRINT " Please start the program via HISTORIC.BAT! "
  COLOR 7, 0
  Pause
  GOTO ProgramEnd
CASE 8      'open the position file
  PRINT "The position file "; CHR$(34); PositionFile$; CHR$(34)
  PRINT "is missing and will be generated."
  ReadingTrial% = ReadingTrial% + 1
  RESUME 6
CASE 61
  PRINT
  PRINT "The program part "; CHR$(34); "COMPUTE.BAS"; CHR$(34); " is";
  PRINT " missing."
  Pause
  RESUME NEXT
CASE 92
  PRINT
  PRINT "The program part "; CHR$(34); "PRINT.BAS"; CHR$(34); " is";
  PRINT " missing."
  Pause
  GOTO ProgramEnd
CASE 101    'open `WorkingPlace$ + InputFileName$`
  RESUME 105    'copy `InputFileName$` to `WorkingPlace$`
CASE ELSE
  Pause
  RESUME NEXT
END SELECT
CASE 62      'Input past end of file
  SELECT CASE ERL
CASE 3
  PRINT "Empty configuration file ";
  PRINT CHR$(34); ConfigurationFile$; CHR$(34); "."
  PRINT "Please load the original files to the current directory."
  Pause
  GOTO ProgramEnd
CASE 4
  RESUME NEXT
CASE 8
  PRINT "Empty position file "; CHR$(34); PositionFile$; CHR$(34); "."
  Pause
  GOTO ProgramEnd
CASE 9
  PRINT "Incomplete position file ";
  PRINT CHR$(34); PositionFile$; CHR$(34); "."
```

```
        Pause
        GOTO ProgramEnd
CASE 100
    Default& = 1
    DataLine$ = Buffer$
    RESUME NEXT
CASE ELSE
    Pause
    RESUME NEXT
END SELECT
CASE 70                                'Permission denied
    SELECT CASE ERL
    CASE 1                            'Writings test to `WorkingPlace$`
        PRINT "The working place directory " + WorkingPlace$
        PRINT "is write protected. Its value is fixed via the DOS command"
        PRINT CHR$(34); "SET HISTORICTEMP=.." + CHR$(34) + "."
        Pause
        SYSTEM                        'Direct program termination
    CASE ELSE
        Pause
        RESUME NEXT
    END SELECT
CASE 75                                'Path/File access error
    SELECT CASE ERL
    CASE 1                            'Writing test to `WorkingPlace$`
        RESUME NEXT
    CASE 102                          '`InputFileName$` being write protected
        RESUME 104
    CASE ELSE
        Pause
        RESUME NEXT
    END SELECT
CASE 76                                'Path not found
    SELECT CASE ERL
    CASE 1
        PRINT "The chosen working directory " + WorkingPlace$
        PRINT " was not found. The program will be terminated."
        Pause
        SYSTEM
    CASE ELSE
        Pause
        RESUME NEXT
    END SELECT
CASE 100
    PRINT "Missing data structure in "; CHR$(34); "HISTORIC.POS"; CHR$(34);
    PRINT "."
    Pause
    GOTO ProgramEnd
CASE 106
    PRINT "The year number is too great. Four digits should be enough."
    Pause
    RESUME NEXT
CASE 110
    PRINT "Senseless dating with ordinary number chaos."
    Pause
    GOTO ProgramEnd
CASE 111
    PRINT "Unknown structure cannot be changed."
```

```

    Pause
    GOTO ProgramEnd
END SELECT
ON ERROR GOTO 0
GOTO ProgramEnd
' _____ END OF THE ERROR HANDLING _____ '

```

```

' =====
' | SUBROUTINES VIA GOSUB |
' =====

```

```

' =====
C0:      'Preparation:
' =====
' Gets all data for the main menu.
'
' Handling:
' 8/18/2001 - 8/19/2001: Norbert Südland, Adelshofen
' 8/31/2002 - 9/ 6/2002: Norbert Südland, Aalen
' 11/22/2006 - 5/ 4/2007: Norbert Südland, Aalen
' Check:
' 8/31/2002 - 9/ 6/2002: Norbert Südland, Aalen
' Translation:
' 9/14/2007:          Norbert Südland, Aalen
' -----
SCREEN 0, 0
COLOR 7, 0

' The `EmptyCharacter$` in principal can be `CHR$(32)` or `CHR$(255)`,
' thus the following assumption owns a general advantage in comparison
' to a constant `EmptyCharacter$ = " "`:
' -----
EmptyCharacter$ = SPACE$(1)

' As `WorkingPlace$` the working directory is used, which is given by
' %HISTORICTEMP%, %TEMP%, or %TMP%.
' If (on old DOS versions) no `WorkingPlace$` is set, then there is a
' trial to write to the data medium that also contains the program.
' Eventually the program will terminate, if the `WorkingPlace$` is not
' writable:
' -----
WorkingPlace$ = ENVIRON$("HISTORICTEMP")
IF WorkingPlace$ = "" THEN
    WorkingPlace$ = ENVIRON$("QBASICTEMP")
    IF WorkingPlace$ <> "" THEN WorkingPlace$ = WorkingPlace$ + "\"
1   MKDIR WorkingPlace$ + "HISTORIC.TMP"
    WorkingPlace$ = WorkingPlace$ + "HISTORIC.TMP\"
ELSE
    IF WorkingPlace$ <> "" THEN WorkingPlace$ = WorkingPlace$ + "\"
END IF

' Check Writability of `WorkingPlace$`:
' -----
BSAVE WorkingPlace$ + "HISTORIC.CHK", 0, 0
KILL WorkingPlace$ + "HISTORIC.CHK"

' Open Configuration File:

```

```

'-----'
InputSuccess% = 0
WHILE InputSuccess% < 2 AND HistoricChoice$ = ""
    Configuration% = FREEFILE
    IF InputSuccess% = 0 THEN
        Buffer$ = WorkingPlace$ + ConfigurationFile$
    ELSE
        Buffer$ = ConfigurationFile$
    END IF
2    OPEN Buffer$ FOR INPUT AS #Configuration%

    'Read in the `Century`:
    '-----'
3    LINE INPUT #Configuration%, Century$

    'Read in Further Configurations:
    '-----'
4    LINE INPUT #Configuration%, InputFileName$
    LINE INPUT #Configuration%, HistoricChoice$
    LINE INPUT #Configuration%, Text$
    IF CountingMode% < 1 OR CountingMode% > 4 THEN
        CountingMode% = VAL(Text$)
    END IF
    CLOSE #Configuration%
    InputSuccess% = InputSuccess% + 1
WEND

'Check, Update, or Set Default Values:
'-----'
Ctry% = VAL(RIGHT$(Century$, 3))
IF Ctry% > 100 OR Ctry% < -10 THEN Ctry% = 0
IF Ctry% = 0 OR LEN(ENVIRON$("CENTURY")) = 4 THEN
    Ctry% = VAL(LEFT$(ENVIRON$("CENTURY"), 1) + "1")
    Ctry% = Ctry% * VAL(RIGHT$(ENVIRON$("CENTURY"), 3))
    IF Ctry% > 100 OR Ctry% < -10 OR Ctry% = 0 THEN
        Ctry% = 21          'The 0th century does not exist!
    END IF
END IF
IF CountingMode% < 1 OR CountingMode% > 4 THEN
    CountingMode% = 1
END IF

'Care About Specialities of WindowsNT:
'-----'
Century$ = FillUpText$(STR$(Ctry% * SGN(Ctry%)), -3, EmptyCharacter$)
IF SGN(Ctry%) = -1 THEN
    Century$ = "-" + Century$
ELSE
    Century$ = "+" + Century$
END IF
5    ENVIRON "CENTURY=" + Century$    'Trial of updating

InputFileName$ = UCASE$(InputFileName$)
InputFileName$ = LEFT$(InputFileName$, STRLEN$(InputFileName$, ".")
IF InputFileName$ = "" THEN InputFileName$ = "TEST"
InputFileName$ = FillUpText$(InputFileName$, 8, EmptyCharacter$) + ".HQL"

HistoricChoice% = VAL(HistoricChoice$)

```

```

IF HistoricChoice% < 1 OR HistoricChoice% > 9 THEN
    HistoricChoice% = 1
END IF

GOSUB SaveConfiguration

'Calculate the Week Day:
'-----'
Ctry% = VAL(Century$)
y% = VAL(RIGHT$(DATE$, 2))
IF y% = 0 THEN
    y% = 100          'Value area for 'y%': 1 to 100
END IF
IF Ctry% = 100 AND y% = 100 THEN Ctry% = 96      'set the same week
IF Ctry% = -10 AND y% = 1 THEN Ctry% = -3      'days for the date
m% = VAL(LEFT$(DATE$, 2))
d% = VAL(MID$(DATE$, 4, 2))
GlobWeekDay% = WeekDay%(Ctry%, y%, m%, d%)
m$ = FillUpText$(LTRIM$(STR$(m%)), -2, EmptyCharacter$)
d$ = FillUpText$(LTRIM$(STR$(d%)), -2, EmptyCharacter$)
y$ = FillUpText$(LTRIM$(STR$(y%)), -4, EmptyCharacter$)
GlobalDate$ = m$ + "/" + d$ + "/" + y$

PRINT "Read in struct positions from HISTORIC.POS ..."
ReadingTrial% = 1
6  SELECT CASE ReadingTrial%
CASE 1
    PositionFile$ = "HISTORIC.POS"
CASE 2
    PositionFile$ = WorkingPlace$ + PositionFile$
CASE 3
    Configuration% = FREEFILE
    OPEN "HISTORIC.STR" FOR INPUT AS #Configuration%
    CLOSE #Configuration%
7  OPEN WorkingPlace$ + "STRUCT.CFG" FOR OUTPUT AS #Configuration%
    PRINT #Configuration%, "HISTORIC.STR"      'Struct file
    PRINT #Configuration%, PositionFile$      'The very position file
    PRINT #Configuration%, "HISTORIC.BAS"      'Return program
    CLOSE #Configuration%
    CHAIN "STRUCT.BAS"
CASE ELSE
    Pause                                     'Programming mistake!
END SELECT

Position% = FREEFILE
8  OPEN PositionFile$ FOR INPUT AS #Position%
9  LINE INPUT #Position%, Text$
    Length% = STRLEN$(Text$, EmptyCharacter$)
    GlobVarNumber% = VAL(LEFT$(Text$, Length%))
    PRINT LTRIM$(STR$(GlobVarNumber%)); " struct elements"
    REDIM SHARED GVBegin%(1 TO GlobVarNumber%)      'AS INTEGER
    REDIM SHARED GVLength%(1 TO GlobVarNumber%)     'AS INTEGER
    REDIM SHARED GVName$(1 TO GlobVarNumber%)       'AS STRING
    REDIM SHARED GVType$(1 TO GlobVarNumber%)       'AS STRING
    FOR c% = 1 TO GlobVarNumber%
        INPUT #Position%, GVBegin%(c%)
        INPUT #Position%, GVLength%(c%)
        INPUT #Position%, GVName$(c%)

```

```

        INPUT #Position%, GVType$(c&)
    NEXT c&
    CLOSE Position%

    DataLength% = SIZEOF$("data")
    IF DataLength% <= 0 THEN ERROR 100

RETURN 'C0 Preparation _____'

'=====
SaveConfiguration:
'=====

'Save Current Configuration:
'-----'
Configuration% = FREEFILE
OPEN WorkingPlace$ + ConfigurationFile$ FOR OUTPUT AS #Configuration%
    PRINT #Configuration%, Century$
    PRINT #Configuration%, UCASE$(InputFileName$)
    PRINT #Configuration%, HistoricChoice%
    PRINT #Configuration%, CountingMode%
CLOSE #Configuration%

RETURN 'SaveConfiguration _____'

'=====
C1:      'Change the Current Date:
'=====
' Will change the "current" date by condidering the historic calendar
' changes on February 29th, 45 before Christ, and on October 15th, 1582.
' It will be demonstrated, that the "year 2000 problem" can be solved much
' more detailed already by DOS and BASIC.
'
' Handling:
' 9/ 6/2002: Norbert Südland, Aalen
' 11/22/2006: Norbert Südland, Aalen
' Check:
'
' Translation:
' 9/14/2007: Norbert Südland, Aalen
'-----'

m$ = LEFT$(GlobalDate$, 2)
d$ = MID$(GlobalDate$, 4, 2)
Year$ = RIGHT$(GlobalDate$, 4)
Length% = LEN(CurrentDate$)
ly% = INT(40! - Length% / 2) + Length% - 9
10 DO
    LOCATE 8, ly%
    result$ = Editor$(m$, 2, 1, InputSuccess%)
    IF InputSuccess% = 1 THEN
        m% = VAL(result$)
    END IF
    LOOP WHILE (m% < 1 OR m% > 12) AND InputSuccess% = 1
    IF InputSuccess% = 0 THEN
        GOTO SetDate
    END IF
    m$ = FillUpText$(LTRIM$(result$), -2, EmptyCharacter$)
    PRINT m$;

```

```

DO
  LOCATE 8, ly% + 3
  result$ = Editor$(d$, 2, 1, InputSuccess%)
  IF InputSuccess% = 1 THEN
    d% = VAL(result$)
  END IF
  LOOP WHILE (d% < 1 OR d% > 31) AND InputSuccess% = 1
  IF InputSuccess% = 0 THEN
    GOTO SetDate
  END IF
  d$ = FillUpText$(LTRIM$(result$), -2, EmptyCharacter$)
  PRINT d$;
DO
  LOCATE 8, ly% + 6
  result$ = Editor$(Year$, 4, 1, InputSuccess%)
  IF InputSuccess% = 1 THEN
    Year% = VAL(result$)
  END IF
  LOOP UNTIL Year% <> 0 OR InputSuccess% = 0
  IF InputSuccess% = 0 THEN
    GOTO SetDate
  END IF
  Year$ = result$
  PRINT Year$;
'=====
SetDate:
'=====
  Year% = VAL(Year$)
  IF Year% > 0 THEN
    Ctry% = INT((Year% - 1) / 100) + 1          'Value area > 0
    y% = ((Year% - 1) MOD 100) + 1             'Value area: 1-100
  ELSEIF Year% < 0 THEN
    Ctry% = INT(Year% / 100)                   'Value area < 0
    y% = 1 + Year% - 100 * Ctry%               'Value area: 1-100
  END IF

  'Gregorianic Calendar Reform:
  '-----'
  Year% = VAL(Year$)
  IF d% = 29 AND m% = 2 AND Year% > 1582 AND Year% MOD 100 = 0 THEN
12   IF Year% MOD 400 <> 0 THEN ERROR 5
  END IF
13   IF Year% = 1582 AND m% = 10 AND d% < 15 AND d% > 4 THEN ERROR 5

  'Julian Calendar:
  '-----'
14   IF d% = 29 AND m% = 2 AND Year% < -45 THEN ERROR 5
  y$ = RIGHT$(STR$(y% + 100), 2)
  IF y% MOD 100 < 80 THEN
    DateValue$ = LTRIM$(m$) + "/" + LTRIM$(d$) + "/20" + y$
  ELSE
    DateValue$ = LTRIM$(m$) + "/" + LTRIM$(d$) + "/19" + y$
  END IF
15   DATE$ = DateValue$
  GlobalDate$ = m$ + "/" + d$ + "/" + Year$

  'Calculate and Show the Week Day (This Function will Change y%):
  '-----'

```

```

GlobWeekDay% = WeekDay%(Ctry%, y%, m%, d%)

LOCATE 8, 1, 0
PRINT SPACE$(80);
Present 1, CurrentDate$, "", 0, "C", 1, 1, 8

'Check Current Century in the Environmental Variables:
'-----'
Century$ = FillUpText$(STR$(Ctry% * SGN(Ctry%)), -3, EmptyCharacter$)
IF SGN(Ctry%) = -1 THEN
    Century$ = "-" + Century$
ELSE
    Century$ = "+" + Century$
END IF
11 IF ENVIRON$("CENTURY") <> Century$ THEN ERROR 7
IF LEN(ENVIRON$("CENTURY")) = 4 THEN ENVIRON "CENTURY=" + Century$

'Save Current Century at the Configuration:
'-----'
GOSUB SaveConfiguration
RETURN 'C1 Change the Current Date _____'

'=====
C2:      'Begin /Enlarge the Data File:
'=====
' Translation: 9/14/2007: Norbert Südland, Aalen
'-----'
GOSUB FileName
IF Success% = 0 THEN RETURN
StopFlag% = 0
InputSuccess% = 1
WHILE GIL% < 16777215 AND StopFlag% = 0 AND InputSuccess% = 1
    GIL% = GIL% + 1
    Data$ = SPACE$(SIZEOF$("data") - 3) + CHR$(179) + CHR$(13) + CHR$(10)
    OK% = SetUp%(Data$, GIL%, HistoricChoice%, InputSuccess%)
    IF InputSuccess% = 0 THEN
        GIL% = GIL% - 1
    ELSE
        IF OK% = 0 THEN
            GIL% = GIL% - 1
        ELSE
            WriteTo InputFile%, DataLength%, GIL%, Data$
        END IF
    END IF
    IF Question$("Close the Data File?", "N", InputSuccess%) = "Y" THEN
        IF InputSuccess% = 1 THEN
            StopFlag% = 1
        END IF
    ELSE
        StopFlag% = 0
    END IF
WEND
CLOSE #InputFile%
RETURN 'C2 Begin /Enlarge the Data File _____'

'=====
C3:      'Delete a Data Record:
'=====

```

```

' Translation:
' 9/14/2007: Norbert Südland, Aalen
'-----
GOSUB FileName
IF Success% = 0 THEN RETURN
DO
    Choice& = Overview&(HistoricChoice%, Choice&, InputSuccess%)
    IF InputSuccess% = 1 THEN GOSUB DeleteRecord
LOOP UNTIL InputSuccess% = 0
CLOSE #InputFile%
RETURN 'C3 Delete a Data Record _____

'=====
C4:      'Pre-Date a Data Record:
'=====
' Translation:
' 9/14/2007: Norbert Südland, Aalen
'-----
GOSUB FileName
IF Success% = 0 THEN RETURN
DO
    Choice& = Overview&(HistoricChoice%, Choice&, InputSuccess%)
    IF InputSuccess% > 0 THEN
        c& = Dating&(HistoricChoice%, Choice&, Term$)
    END IF
LOOP UNTIL InputSuccess% = 0
CLOSE #InputFile%
RETURN 'C4 Pre-Date a Data Record _____

'=====
C5:      'Overview and Correction:
'=====
' Translation:
' 9/14/2007: Norbert Südland, Aalen
'-----
GOSUB FileName
IF Success% = 0 THEN RETURN
InputSuccess% = 1
DO
    Choice& = Overview&(HistoricChoice%, Choice&, InputSuccess%)
    IF InputSuccess% > 0 THEN
        IF Question$("Show Pre-datings?", "N", InputSuccess%) = "Y" THEN
            IF InputSuccess% = 1 THEN
                c& = Dating&(HistoricChoice%, Choice&, Term$)
            END IF
        ELSE
            IF InputSuccess% = 1 THEN
                Data$ = Load$(InputFile%, DataLength%, Choice&)
                OK% = SetUp$(Data$, Choice&, HistoricChoice%, InputSuccess%)
                IF InputSuccess% = 1 THEN
                    IF OK% = 0 THEN
                        GOSUB DeleteRecord
                    ELSE
                        Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
                        WriteTo InputFile%, DataLength%, Choice&, Data$
                    END IF
                END IF
            END IF
        END IF
    END IF
END IF

```

```

        InputSuccess% = 1      'Avoid too deep return jump by [ESC].
    END IF
END IF
    LOOP UNTIL InputSuccess% = 0
CLOSE #InputFile%
RETURN 'C5 Overview and Correction _____'

'=====
C6:      'Calculation:
'=====
'Translation:
'  9/14/2007: Norbert Südland, Aalen
'-----
GOSUB FileName
IF Success% = 0 THEN RETURN
IF GIL& > 0 THEN
    New% = 0
    IF Question$("New Calculation for All Data?", "N", Success%) = "Y" THEN
        IF Success% = 1 THEN New% = 1
    END IF
    IF Success% = 1 THEN
        LOCATE 25, 1
        PRINT SPACE$(79);
        Text$ = "Which Calculation Method? [1-4]"
        IF CountingMode% >= 1 AND CountingMode% <= 4 THEN
            Template$ = LTRIM$(STR$(CountingMode%))
        ELSE
            Template$ = "1"
        END IF
        Present 2, Text$, Template$, -1, "C", 1, 1, 25
        x% = CSRLIN
        ly% = POS(0) - 1
        DO
            LOCATE x%, ly%
            CountingMode% = VAL(Editor$(Template$, 1, 2, Success%))
        LOOP UNTIL CountingMode% >= 1 AND CountingMode% <= 4 OR Success% = 0
        IF Success% = 1 THEN
            PRINT LTRIM$(STR$(CountingMode%));
        END IF
    END IF
    IF Success% = 1 THEN
        Configuration% = FREEFILE
        OPEN WorkingPlace$ + "COMPUTE.CFG" FOR OUTPUT AS #Configuration%
        PRINT #Configuration%, UCASE$(InputFileName$)
        PRINT #Configuration%, GIL&
        PRINT #Configuration%, New%
        PRINT #Configuration%, CountingMode%
        PRINT #Configuration%, "HISTORIC.BAS" 'Return program
        CLOSE #Configuration%
    END IF
END IF
CLOSE #InputFile%
IF GIL& > 0 AND Success% = 1 THEN
    GOSUB SaveConfiguration
61 CHAIN "COMPUTE.BAS"
END IF
RETURN 'C6 Calculation _____'

```

```

'=====
C7:      'Sort the Data File:
'=====
'  Enables a manual sorting like in a data box:
'
'  Handling:
'  12/18/2002: Norbert Südland, Aalen
'  Translation:
'  9/14/2007: Norbert Südland, Aalen
'-----
GOSUB FileName
IF Success% = 0 THEN RETURN
  InputSuccess% = 1
  DO
    Choice& = Overview&(6, Choice&, InputSuccess%)
    IF InputSuccess% > 0 THEN
      Data$ = Load$(InputFile%, DataLength%, Choice&)
      Target& = Overview&(7, Choice&, InputSuccess%)
      IF InputSuccess% > 0 THEN
        IF Choice& <= Target& THEN
          FOR c& = Choice& TO Target& - 1&
            SwapTerm$ = Load$(InputFile%, DataLength%, c& + 1)
            Change SwapTerm$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
            WriteTo InputFile%, DataLength%, c&, SwapTerm$
          NEXT c&
        ELSE
          FOR c& = Choice& TO Target& + 1& STEP -1
            SwapTerm$ = Load$(InputFile%, DataLength%, c& - 1)
            Change SwapTerm$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
            WriteTo InputFile%, DataLength%, c&, SwapTerm$
          NEXT c&
        END IF
        Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
        WriteTo InputFile%, DataLength%, Target&, Data$
      END IF
    END IF
  LOOP UNTIL InputSuccess% = 0
  CLOSE #InputFile%
RETURN 'C7 Sort the Data File _____

'=====
C8:      'Show /Transform the Result:
'=====
'  Last Correction:
'  10/ 9/2007: Norbert Südland, Aalen
'  Translation:
'  9/14/2007: Norbert Südland, Aalen
'-----
GOSUB FileName
IF Success% = 0 THEN RETURN
  DO
    Choice& = Overview&(8, Choice&, InputSuccess%)
    IF InputSuccess% = 0 THEN
      Changed% = 0
    ELSE
      Data$ = Load$(InputFile%, DataLength%, Choice&)
      Different% = Dating$(HistoricChoice%, Choice&, Term$)
    END IF
  
```

```

LOOP UNTIL Different% <> 0 OR InputSuccess% = 0
IF LTRIM$(Term%) <> "" THEN
    Comparison$ = Load$(InputFile%, DataLength%, Choice%)
    Change Comparison$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
    WriteTo InputFile%, DataLength%, Choice%, Comparison$

    Variable$ = "data.date[" + LTRIM$(STR$(INT((Different% + 1) / 2))) + "]"
    Summand1$ = Part$(Comparison$, Variable$)
    IF Different% MOD 2 = 1 THEN
        Change Summand1$, "date.minimum", Term$
    ELSE
        Change Summand1$, "date.maximum", Term$
    END IF
    Change Summand1$, "date.status", SPACE$(1)
    NewStatus Summand1$
    Summand2$ = Part$(Data$, Variable$)
    Change Summand2$, "date.status", SPACE$(1)
    NewStatus Summand2$
    Add Summand1$, "U", "-", Summand2$, 1, 1, "-"
    IF Different% MOD 2 = 1 THEN
        Summand2$ = Part$(Summand1$, "date.minimum")
        Change Summand1$, "date.maximum", Summand2$
    ELSE
        Summand2$ = Part$(Summand1$, "date.maximum")
        Change Summand1$, "date.minimum", Summand2$
    END IF
    NewStatus (Summand1$)
    Difference$ = Summand1$
    Length% = SIZEOF$("date")
    IF Difference$ <> SPACE$(Length%) THEN
        Present 2, "Time Offset: " + Difference$, "", 0, "C", 1, 1, 3
        FOR c% = 1 TO GIL%
            Text$ = " Data record " + NameEntry$(c%) + " ("
            Text$ = Text$ + LTRIM$(STR$(c%)) + "/" + LTRIM$(STR$(GIL%))
            Text$ = Text$ + ") is being checked. "
            Present 1, "", Text$, 0, "C", 1, 1, 23
            Data$ = Load$(InputFile%, DataLength%, c%)
            FOR ly% = 1 TO 6
                Summand1$ = Difference$
                Variable$ = "data.date[" + LTRIM$(STR$(ly%)) + "]"
                Summand2$ = Part$(Data$, Variable$)
                IF Summand2$ <> SPACE$(Length%) THEN
                    Status$ = Part$(Summand2$, "date.status")
                    Change Summand2$, "date.status", SPACE$(1)
                    NewStatus Summand2$
                    Add Summand1$, ">", "+", Summand2$, 0, 1, " "
                    IF LTRIM$(Summand1$) = "" THEN
                        ly% = 6
                        c% = GIL%
                    ELSE
                        Change Summand1$, "date.status", Status$
                        Change Data$, Variable$, Summand1$
                    END IF
                END IF
            NEXT ly%
        NEXT c%
    IF LTRIM$(Summand1$) <> "" THEN
        FOR c% = 1 TO GIL%

```

```

Text$ = " Data record " + NameEntry$(c&) + " ("
Text$ = Text$ + LTRIM$(STR$(c&)) + "/" + LTRIM$(STR$(GIL&))
Text$ = Text$ + ") is being transformed. "
Present 1, "", Text$, 0, "C", 1, 1, 23
Data$ = Load$(InputFile%, DataLength%, c&)
FOR ly% = 1 TO 6
    Summand1$ = Difference$
    Variable$ = "data.date[" + LTRIM$(STR$(ly%)) + "]"
    Summand2$ = Part$(Data$, Variable$)
    IF Summand2$ <> SPACE$(Length%) THEN
        Status$ = Part$(Summand2$, "date.status")
        Change Summand2$, "date.status", SPACE$(1)
        NewStatus Summand2$
        Add Summand1$, "»", "+", Summand2$, 0, 1, " "
        Change Summand1$, "date.status", Status$
        Change Data$, Variable$, Summand1$
    END IF
NEXT ly%
Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
WriteTo InputFile%, DataLength%, c&, Data$
NEXT c&
END IF
ELSE
    Present 2, "Time offset: not available.", "", 0, "C", 1, 1, 3
    Pause
END IF
END IF
CLOSE #InputFile%
RETURN 'C8 Show /Transform the Result _____'

'=====
C9:      'Print the Data File:
'=====
' Translation:
'  9/15/2007: Norbert Südland, Aalen
'-----
GOSUB FileName
IF Success% <> 0 THEN

91 Configuration% = FREEFILE
OPEN WorkingPlace$ + "PRINT.CFG" FOR OUTPUT AS #Configuration%
PRINT #Configuration%, InputFileName$
PRINT #Configuration%, STR$(GIL&)
PRINT #Configuration%, "HISTORIC.BAS"
CLOSE #Configuration%
92 CHAIN "PRINT.BAS"
END IF
RETURN 'C9 Print the Data File _____'

'=====
FileName:
'=====
' Will ask for the file name.
' The overview on the file names must be available independently!
' The overview on the file names can be chosen by the arrow keys.
'
' Handling:
'  1/ 8/2003 - 2/ 8/2003: Norbert Südland

```

```
' 11/14/2006:          Norbert Südland
' Translation:
' 9/15/2007:          Norbert Südland
'-----'
```

```
GOSUB DirectoryOverview
```

```
Success% = -1
```

```
WHILE Success% < 0
```

```
    LOCATE 22, 1
```

```
    PRINT SPACE$(80);
```

```
    LOCATE 23, 1
```

```
    PRINT SPACE$(80);
```

```
InputFile% = FREEFILE
```

```
OPEN WorkingPlace$ + "HQL.DAT" FOR INPUT AS #InputFile%
```

```
    LINE INPUT #InputFile%, Text$
```

```
    c% = 0
```

```
    IF Default% = 0 THEN
```

```
        WHILE EOF(InputFile%) = 0
```

```
            LINE INPUT #InputFile%, DataLine$
```

```
            c% = c% + 1
```

```
        WEND
```

```
        Default% = c%
```

```
    ELSE
```

```
        FOR c% = 1 TO Default%
```

```
100     LINE INPUT #InputFile%, DataLine$
```

```
        IF c% = 1 THEN
```

```
            Buffer$ = DataLine$
```

```
        END IF
```

```
    NEXT c%
```

```
    END IF
```

```
    DataLine$ = LEFT$(DataLine$, 8)
```

```
CLOSE #InputFile%
```

```
LOCATE 23, 1
```

```
PRINT SPACE$(80);
```

```
IF LEN(Text$) > 80 - 16 THEN
```

```
    Text$ = LEFT$(Text$, 80 - 16 - 20) + " ... " + RIGHT$(Text$, 13)
```

```
END IF
```

```
Present 1, "Data Directory: ", Text$, 0, "C", 1, 1, 22
```

```
Present 2, "File Name: ", DataLine$ + ".HQL", 0, "C", 1, 1, 23
```

```
Buffer$ = "[ " + CHR$(25) + " ], [ " + CHR$(24)
```

```
Buffer$ = Buffer$ + " ], [ " + CHR$(17) + CHR$(196) + CHR$(217)
```

```
Buffer$ = Buffer$ + " ], [ Einfg ], [ ESC ]"
```

```
Present 1, "Input Choice: ", Buffer$, 0, "C", 1, 1, 25
```

```
IF DataLine$ = SPACE$(8) THEN
```

```
    Answer$ = CHR$(0) + CHR$(82)
```

```
ELSE
```

```
    Answer$ = KeyInput$
```

```
END IF
```

```
SELECT CASE ASC(Answer$)
```

```
CASE 10, 13
```

```
    InputFileName$ = DataLine$
```

```
    Success% = 1
```

```
CASE 27
```

```
    Success% = 0
```

```
END SELECT
```

```

IF LEN(Answer$) = 2 THEN
    Answer$ = RIGHT$(Answer$, 1)
    SELECT CASE ASC(Answer$)
    CASE 71          'Home
        Default& = 1
    CASE 72, 75      'Cursor up, Cursor left
        Default& = Default& - 1
    CASE 77, 80      'Cursor right, Cursor down
        Default& = Default& + 1
    CASE 79          'End
        Default& = 0
    CASE 82          'Insert
        LOCATE 25, 1
        PRINT SPACE$(80);
        LOCATE 23, 35
        Length% = STRLEN%(DataLine$, ".") + 1
        IF Length% = 1 THEN
            DataLine$ = SPACE$(8)
            Length% = 9
        END IF
        DataLine$ = Editor$(LEFT$(DataLine$, Length% - 1), 8, 2, Success%)
        IF Success% = 0 THEN
            Success% = -1
        END IF
    END SELECT
END IF
WEND
InputFileName$ = UCASE$(DataLine$) + ".HQL"
LOCATE 23, 35
PRINT SPACE$(40)
IF Success% = 0 THEN
    GOTO EndFileName
END IF
LOCATE 23, 18
PRINT "Data file "; InputFileName$; " ";

GOSUB OpenInputFile
IF HistoricChoice% = 2 AND GIL& >= 16777215 THEN
    PRINT "is completely filled."
    Pause
ELSE
    PRINT "owns"; GIL&; "data records."
END IF
IF GIL& = 0 THEN Pause

GOSUB SaveConfiguration
'=====
EndFileName:
'=====
    KILL WorkingPlace$ + "HQL.DAT"
RETURN 'FileName _____

'=====
DirectoryOverview:
'=====
    ' Will generate a standardized list of the directory content.
    '
    ' Handling:

```

```
' 1/ 8/2003 - 2/ 8/2003: Norbert Südland
' 11/14/2006:           Norbert Südland
' 8/11/2009:           Norbert Südland, D-73431 Aalen
' Translation:
' 9/15/2007:           Norbert Südland
'-----'
```

```
CLOSE #InputFile%
Success% = 0
c& = 0
Default& = 0
WHILE Default& = 0 AND Success% < 2

'Will Work Differently at Different DOS Versions:
'-----'

SELECT CASE Success%
CASE 0
    SHELL "DIR " + WorkingPlace$ + " *.* > " + WorkingPlace$ + "DIR.DAT"
CASE 1
    SHELL "DIR *.* > " + WorkingPlace$ + "DIR.DAT"
END SELECT

'Generate List of File Names "*.HQL":
'-----'

DirFile% = FREEFILE
OPEN WorkingPlace$ + "DIR.DAT" FOR BINARY ACCESS READ AS #DirFile%
HQLFile% = FREEFILE
OPEN WorkingPlace$ + "HQL.DAT" FOR OUTPUT AS #HQLFile%
Present 1, "Trying to copy *.HQL files...", "", 0, "C", 1, 1, 22
WHILE EOF(DirFile%) = 0
    LINEINPUT DirFile%, DataLine$
    IF INSTR(2, DataLine$, ":\") THEN
        Found% = INSTR(2, DataLine$, ":\") - 2
        Buffer$ = RIGHT$(DataLine$, LEN(DataLine$) - Found%)
        IF RIGHT$(Buffer$, 2) = "\", THEN
            Buffer$ = LEFT$(Buffer$, LEN(Buffer$) - 2)
        END IF
        PRINT #HQLFile%, Buffer$
    ELSEIF MID$(DataLine$, 9, 4) = " HQL" THEN
        DataLine$ = LEFT$(DataLine$, 8) + ".HQL"
        c& = c& + 1
        PRINT #HQLFile%, DataLine$
        IF DataLine$ = InputFileName$ THEN
            Default& = c&
        END IF
    END IF
    IF Success% = 1 THEN

        'Generate an Unchecked Copy:
        '-----'

        InputFile% = FREEFILE
        OPEN DataLine$ FOR BINARY ACCESS READ AS #InputFile%
        Copy% = FREEFILE
        Buffer$ = WorkingPlace$ + DataLine$
        OPEN Buffer$ FOR BINARY ACCESS WRITE AS #Copy%
        Data$ = SPACE$(SIZEOF("data"))
        c& = 0
        WHILE EOF(InputFile%) = 0
```

```

117      GET #InputFile%, c& * DataLength% + 1, Data$
      Buffer$ = Part$(Data$, "data.name")
      IF ASC(LEFT$(LTRIM$(Buffer$), 1)) <> 0 THEN

          'Adjust Old File Versions to Current File Versions:
          '-----'
          Buffer$ = Part$(Data$, "data.p[1].moment1")
          Change Data$, "data.p[1].moment1", ToFile$(Buffer$)
          Buffer$ = Part$(Data$, "data.p[1].direct")
          Change Data$, "data.p[1].direct", ToFile$(Buffer$)
          Buffer$ = Part$(Data$, "data.p[1].moment2")
          Change Data$, "data.p[1].moment2", ToFile$(Buffer$)

          Buffer$ = Part$(Data$, "data.p[2].moment1")
          Change Data$, "data.p[2].moment1", ToFile$(Buffer$)
          Buffer$ = Part$(Data$, "data.p[2].direct")
          Change Data$, "data.p[2].direct", ToFile$(Buffer$)
          Buffer$ = Part$(Data$, "data.p[2].moment2")
          Change Data$, "data.p[2].moment2", ToFile$(Buffer$)

          Buffer$ = Part$(Data$, "data.r[1].moment1")
          Change Data$, "data.r[1].moment1", ToFile$(Buffer$)
          Buffer$ = Part$(Data$, "data.r[1].moment2")
          Change Data$, "data.r[1].moment2", ToFile$(Buffer$)

          Buffer$ = Part$(Data$, "data.r[2].moment1")
          Change Data$, "data.r[2].moment1", ToFile$(Buffer$)
          Buffer$ = Part$(Data$, "data.r[2].moment2")
          Change Data$, "data.r[2].moment2", ToFile$(Buffer$)

          Buffer$ = Part$(Data$, "data.simultaneous")
          Change Data$, "data.simultaneous", ToFile$(Buffer$)

          Buffer$ = CHR$(179) + CHR$(13) + CHR$(10)
          Change Data$, "data.end", Buffer$

          PUT #Copy%, c& * DataLength% + 1, Data$
      END IF
      c& = c& + 1
  WEND
  CLOSE #Copy%
  Buffer$ = ""
  CLOSE #InputFile%
END IF
END IF
WEND
Present 1, SPACE$(40), "", 0, "C", 1, 1, 22
CLOSE #HQLFile%
CLOSE #DirFile%
KILL WorkingPlace$ + "DIR.DAT"
Success% = Success% + 1
WEND

RETURN 'DirectoryOverview _____'

'=====
OpenInputFile:
'=====

```

```

' Will open the input file.
' Only if the input file in the working directory is empty, in the
' calling directory will be searched for a possibility to copy.
'
' Handling:
' 12/15/2002 - 1/20/2003: Norbert Südland
' Translation:
' 9/15/2007:          Norbert Südland
'-----'
Length% = DataLength%
Data$ = SPACE$(DataLength%)
CopyOfFile$ = WorkingPlace$ + InputFileName$      'Use WorkingPlace$!
GIL% = 0
InputFile% = FREEFILE
101 OPEN CopyOfFile$ FOR BINARY ACCESS READ WRITE AS #InputFile%
    Escape% = 0
    Data$ = SPACE$(DataLength%)
    WHILE Escape% = 0
102     GET #InputFile%, GIL% * DataLength% + 1, Data$
        IF GIL% = 0 THEN GOSUB FileDocumentation
        IF ASC(LEFT$(LTRIM$(Part$(Data$, "data.name")), 1)) <> 0 THEN
            Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
            PUT #InputFile%, GIL% * DataLength% + 1, Data$      'Writing test!
            GIL% = GIL% + 1
        ELSE
            Escape% = 1
        END IF
    WEND
103     GIL% = GIL% - 1
        IF GIL% > 0 THEN RETURN      'The file exists and is writable!
104 CLOSE #InputFile%

'If the file is not writable, a copy is generated:
'-----'
GIL% = 0
InputFile% = FREEFILE
105 OPEN InputFileName$ FOR BINARY ACCESS READ AS #InputFile%
    Copy% = FREEFILE
106 OPEN CopyOfFile$ FOR BINARY ACCESS WRITE AS #Copy%
    Escape% = 0
    Data$ = SPACE$(DataLength%)
    WHILE EOF(InputFile%) = 0 AND Escape% = 0
107     GET #InputFile%, GIL% * DataLength% + 1, Data$
        IF GIL% = 0 THEN GOSUB FileDocumentation
        IF ASC(LEFT$(LTRIM$(Part$(Data$, "data.name")), 1)) <> 0 THEN

            'Adjust Old File Versions to Current File Versions:
            '-----'
            Buffer$ = Part$(Data$, "data.p[1].moment1")
            Change Data$, "data.p[1].moment1", ToFile$(Buffer$)
            Buffer$ = Part$(Data$, "data.p[1].direct")
            Change Data$, "data.p[1].direct", ToFile$(Buffer$)
            Buffer$ = Part$(Data$, "data.p[1].moment2")
            Change Data$, "data.p[1].moment2", ToFile$(Buffer$)

            Buffer$ = Part$(Data$, "data.p[2].moment1")
            Change Data$, "data.p[2].moment1", ToFile$(Buffer$)
            Buffer$ = Part$(Data$, "data.p[2].direct")

```

```

        Change Data$, "data.p[2].direct", ToFile$(Buffer$)
        Buffer$ = Part$(Data$, "data.p[2].moment2")
        Change Data$, "data.p[2].moment2", ToFile$(Buffer$)

        Buffer$ = Part$(Data$, "data.r[1].moment1")
        Change Data$, "data.r[1].moment1", ToFile$(Buffer$)
        Buffer$ = Part$(Data$, "data.r[1].moment2")
        Change Data$, "data.r[1].moment2", ToFile$(Buffer$)

        Buffer$ = Part$(Data$, "data.r[2].moment1")
        Change Data$, "data.r[2].moment1", ToFile$(Buffer$)
        Buffer$ = Part$(Data$, "data.r[2].moment2")
        Change Data$, "data.r[2].moment2", ToFile$(Buffer$)

        Buffer$ = Part$(Data$, "data.simultaneous")
        Change Data$, "data.simultaneous", ToFile$(Buffer$)

        Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)

108      PUT #Copy%, GIL& * DataLength% + 1, Data$      'Writing test!
        GIL& = GIL& + 1
    ELSE
109      Escape% = 1
    END IF
    WEND
    GIL& = GIL& - 1
    CLOSE #Copy%
    CLOSE #InputFile%
    OPEN CopyOfFile$ FOR BINARY ACCESS READ WRITE AS #InputFile%
RETURN 'OpenInputFile _____'

'=====
DeleteRecord:
'=====
' Will delete a file record:
'
' Handling:
'   1/21/2003:          Norbert Südland
' Check:
' 12/15/2002 - 1/21/2003: Norbert Südland
' Translation:
'   9/15/2007:          Norbert Südland
'-----

IF Choice& <> 0 THEN
    Text$ = "Record " + NameEntry$(Choice&) + "is being deleted."
    Present 1, Text$, "", 0, "C", 1, 1, 24

    'Generate Intermediate Copy:
    '-----
    CopyOfFile$ = LEFT$(InputFileName$, STRLEN$(InputFileName$, ".") + ".CPY"
    CopyOfFile$ = WorkingPlace$ + CopyOfFile$
    Copy% = FREEFILE
    Length% = DataLength%
    OPEN CopyOfFile$ FOR BINARY ACCESS READ WRITE AS #Copy%
        Data$ = SPACE$(DataLength%)
        GOSUB FileDocumentation
        PUT #Copy%, 1, Data$
        FOR c& = 1 TO GIL&

```

```

        Data$ = Load$(InputFile%, DataLength%, c&)
        Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
        WriteTo Copy%, DataLength%, c&, Data$
    NEXT c&
CLOSE #InputFile%

'Delete Original:
'-----'
KILL WorkingPlace$ + InputFileName$

'Generate New File Containing Less Records:
'-----'
CopyOfFile$ = WorkingPlace$ + InputFileName$           'Use WorkingPlace$!
OPEN CopyOfFile$ FOR BINARY ACCESS READ WRITE AS #InputFile%
Data$ = SPACE$(DataLength%)
GOSUB FileDocumentation
PUT #InputFile%, 1, Data$
Offset% = 0
FOR c& = 1 TO GIL&
    IF c& = Choice& THEN
        Offset% = 1
    ELSE
        Data$ = Load$(Copy%, DataLength%, c&)
        Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
        WriteTo InputFile%, DataLength%, c& - Offset%, Data$
    END IF
NEXT c&
GIL& = GIL& - Offset%
CLOSE #Copy%

'Delete Intermediate File:
'-----'
CopyOfFile$ = LEFT$(InputFileName$, STRLEN$(InputFileName$, ".") + ".CPY"
CopyOfFile$ = WorkingPlace$ + CopyOfFile$
KILL CopyOfFile$
END IF
RETURN 'DeleteRecord _____'

'=====
FileDocumentation:
'=====
' Will generate a documentation line for the 0th data record.
'
' Handling:
' 1/21/2003: Norbert Suedland
' Translation:
' 9/15/2007: Norbert Suedland
'-----'
FOR x% = 1 TO 6
    Buffer$ = "Minimum[" + LTRIM$(STR$(x%)) + "]"
    Change Data$, "data.date[" + LTRIM$(STR$(x%)) + "].minimum", Buffer$
    Buffer$ = "Maximum[" + LTRIM$(STR$(x%)) + "]"
    Change Data$, "data.date[" + LTRIM$(STR$(x%)) + "].maximum", Buffer$
NEXT x%
Change Data$, "data.name", "Name /Event"
Change Data$, "data.source", "1st Source"
FOR x% = 1 TO 2
    Buffer$ = OrdinaryNumber$(x%) + " Predecessor"

```

```

Change Data$, "data.p[" + LTRIM$(STR$(x%)) + "]", Buffer$
Buffer$ = OrdinaryNumber$(x%) + " Date"
Change Data$, "data.r[" + LTRIM$(STR$(x%)) + "].date", Buffer$
Buffer$ = OrdinaryNumber$(x%) + " Relation Name"
Change Data$, "data.r[" + LTRIM$(STR$(x%)) + "].name", Buffer$
Buffer$ = OrdinaryNumber$(x%) + " Duration"
Change Data$, "data.r[" + LTRIM$(STR$(x%)) + "].duration", Buffer$
Buffer$ = OrdinaryNumber$(x% + 1) + " Source"
Change Data$, "data.r[" + LTRIM$(STR$(x%)) + "].source", Buffer$
NEXT x%
Change Data$, "data.end", CHR$(179) + CHR$(13) + CHR$(10)
RETURN 'FileDocumentation _____'

```

SUBROUTINES AND FUNCTIONS

```

=====
SUB Add (Summand1$, Kind$, Direction$, Default$, ONum1%, ONum2%, NextSt$)
=====

```

```

' Will add dates in the correct way.

```

```

' Handling:

```

```

' 3/27/2003: Norbert Südland

```

```

' Translation:

```

```

' 9/15/2007: Norbert Südland
-----

```

```

DIM Summand2$
DIM status1$
DIM status2$
DIM Status$
DIM term1$
DIM term2$
DIM Length%
DIM s0%, s1%, s2%, s3%, s4%, s5%, s6%
DIM y0%, y1%, y2%, y3%, y4%, y5%, y6%
DIM m0%, m1%, m2%, m3%, m4%, m5%, m6%
DIM d0%, d1%, d2%, d3%, d4%, d5%, d6%
DIM start%
DIM finish%
DIM c%

```

```

'Preparation:
'-----

```

```

Summand2$ = Default$
NewStatus Summand1$
status1$ = Part$(Summand1$, "date.status")
NewStatus Summand2$
status2$ = Part$(Summand2$, "date.status")
Length% = SIZEOF$("date")
IF Summand1$ = SPACE$(Length%) AND Summand2$ = SPACE$(Length%) THEN
    GOTO EndAdding
END IF

```

```

'Care for Kind of Addition:
'-----

```

```

IF Kind$ = "U" THEN      'Union set
  Status$ = status1$
  IF Status$ <> status2$ THEN
    Status$ = " "
  END IF
ELSE
  IF Kind$ = ">>" THEN    'Summand2$ is dominant
    Status$ = status2$
    IF Summand2$ = SPACE$(Length%) THEN
      Summand1$ = Summand2$
      GOTO EndAdding
    END IF
  ELSE
    GOTO EndAdding
  END IF
END IF

'The very Adding:
'-----'
TermToNumber Part$(Summand1$, "date.minimum"), s3%, y3%, m3%, d3%
TermToNumber Part$(Summand2$, "date.minimum"), s4%, y4%, m4%, d4%
TermToNumber Part$(Summand1$, "date.maximum"), s5%, y5%, m5%, d5%
TermToNumber Part$(Summand2$, "date.maximum"), s6%, y6%, m6%, d6%
IF ONum1% <> 0 THEN
  TimeShift s3%, y3%, m3%, d3%, status1$, s5%, y5%, m5%, d5%, "--"
END IF
IF ONum2% <> 0 THEN
  TimeShift s4%, y4%, m4%, d4%, status2$, s6%, y6%, m6%, d6%, "--"
END IF
Reverse s4%, y4%, m4%, d4%, s6%, y6%, m6%, d6%, Direction$
SELECT CASE Status$
CASE CHR$(242)          '>='
  start% = 1
  finish% = 1
CASE CHR$(243)          '<='
  start% = 2
  finish% = 2
CASE " "
  start% = 1
  finish% = 2
END SELECT
FOR c% = start% TO finish%
  IF c% = 1 THEN
    s1% = s3%: y1% = y3%: m1% = m3%: d1% = d3%
    s2% = s4%: y2% = y4%: m2% = m4%: d2% = d4%
  ELSE
    s1% = s5%: y1% = y5%: m1% = m5%: d1% = d5%
    s2% = s6%: y2% = y6%: m2% = m6%: d2% = d6%
  END IF
  d0% = d1% + d2%
  m0% = m1% + m2%
  y0% = s1% * y1% + s2% * y2%
  s0% = 1
  RestSystem s0%, y0%, m0%, d0%
  IF ABS(y0%) >= 10000 THEN
    c% = finish%
  ELSE
    IF c% = 1 THEN

```

```

        s3% = s0%: y3% = y0%: m3% = m0%: d3% = d0%
    ELSE
        s5% = s0%: y5% = y0%: m5% = m0%: d5% = d0%
    END IF
END IF
NEXT c%
IF ABS(y0%) >= 10000 THEN GOTO EndAdding
IF (ONum1% <> 0 OR ONum2% <> 0) AND NextSt$ <> "-" THEN
    TimeShift s3%, y3%, m3%, d3%, status1$, s5%, y5%, m5%, d5%, "+"
END IF
SELECT CASE Status$
CASE CHR$(242) ' >=
    term1$ = NumberToTerm$(s3%, y3%, m3%, d3%)
    term2$ = SPACE$(SIZEOF$("term"))
CASE CHR$(243) ' <=
    term1$ = SPACE$(SIZEOF$("term"))
    term2$ = NumberToTerm$(s5%, y5%, m5%, d5%)
CASE " "
    term1$ = NumberToTerm$(s3%, y3%, m3%, d3%)
    term2$ = NumberToTerm$(s5%, y5%, m5%, d5%)
END SELECT
Change Summand1$, "date.minimum", term1$
Change Summand1$, "date.status", Status$
Change Summand1$, "date.maximum", term2$

'====='
EndAdding:
'====='
    IF ABS(y0%) >= 10000 THEN
        Summand1$ = SPACE$(SIZEOF$("date"))
    END IF
END SUB 'Add _____'

'=====
SUB Change (Data$, Variable$, Contents$)
'=====
' Will put 'Contents$' into the correct position of 'Data$'.
'
' Handling:
' 9/ 4/2001 - 1/21/2003: Norbert Suedland
' Translation:
' 10/ 6/2007: Norbert Suedland
'-----

DIM seek$
DIM Length%
DIM p% 'position
DIM begin%
DIM finish%
DIM where%
DIM np% 'next position
DIM i% 'index

Length% = LEN(Variable$)
p% = STRLEN$(Variable$, ".")
IF p% = Length% THEN
    seek$ = Variable$
ELSE

```

```

    p% = p% + 1
    p% = p% + STRLEN%(MID$(Variable$, p% + 1, Length% - p%), ".")
    IF p% < Length% THEN
        seek$ = LEFT$(Variable$, p%)
    ELSE
        seek$ = Variable$
    END IF
END IF
begin% = 1
DO
    finish% = GlobVarNumber%
    where% = Number%(seek$, 1, finish%)
    IF where% = 0 THEN
        STOP 'seek$ not found!
        ERROR 111
    END IF
    begin% = begin% + GVBEGIN%(where%) - 1
    IF p% < Length% THEN
        p% = p% + 1
        np% = STRLEN%(MID$(Variable$, p% + 1, Length% - p%), ".")
        seek$ = GVType$(where%) + MID$(Variable$, p%, np% + 1)
        p% = p% + np%
    ELSE
        p% = Length% + 1
    END IF
LOOP WHILE p% <= Length%

i% = LEN(Contents$)
IF i% < GVLength%(where%) THEN
    Contents$ = SPACE$(GVLength%(where%) - i%) + Contents$
ELSE
    IF i% > GVLength%(where%) THEN
        Contents$ = LEFT$(Contents$, GVLength%(where%))
    END IF
END IF

MID$(Data$, begin%, GVLength%(where%)) = Contents$
END SUB 'Change _____

'=====
FUNCTION CurrentDate$
'=====
' Will generate the presentation line containing the current date.
'
' Handling:
' 9/ 4/2001: Norbert Suedland, Munich
' Translation:
' 9/15/2007: Norbert Suedland, Aalen
'-----
DIM Text$

SELECT CASE GlobWeekDay%
CASE 1
    Text$ = "Sunday"
CASE 2
    Text$ = "Monday"
CASE 3

```

```

        Text$ = "Tuesday"
    CASE 4
        Text$ = "Wednesday"
    CASE 5
        Text$ = "Thursday"
    CASE 6
        Text$ = "Friday"
    CASE 7
        Text$ = "Saturday"
END SELECT
Text$ = Text$ + " (" + OrdinaryNumber$(GlobWeekDay%)
Text$ = Text$ + " day of the week), " + GlobalDate$

CurrentDate$ = Text$
END FUNCTION 'CurrentDate$ _____'

'=====
FUNCTION Dating% (HistoricChoice%, Choice&, Term$)
'=====
'
' Handling:
' 8/ 5/2001 - 9/ 4/2001:    Norbert Südland, Munich
' 12/16/2002 - 3/27/2003:   Norbert Südland, Aalen
' Translation:
' 10/ 6/2007:              Norbert Südland, Aalen
'-----
DIM Text$      'AS STRING
DIM Data$
DIM a$
DIM Variable$
DIM DateValue$
DIM y$
DIM m$
DIM d$
DIM minimum$
DIM maximum$
DIM Status$
DIM source$
DIM Buffer$
DIM finish%    'AS INTEGER
DIM x%
DIM ly%
DIM c%         'counter
DIM current%
DIM Answer%
DIM NextSt%
DIM InputSuccess%
DIM repetition%
DIM Length%
DIM AddSource%

CLS
Term$ = ""
IF Choice& >= 100 THEN
    Text$ = LTRIM$(STR$(INT(Choice& / 100)))
    IF INT((Choice& MOD 100) / 10) = 0 THEN
        Text$ = Text$ + "0"

```

```
END IF
ELSE
  Text$ = ""
END IF
Text$ = Text$ + OrdinaryNumber$(Choice& MOD 100) + " File Record "
Text$ = Text$ + NameEntry$(Choice&)
SELECT CASE HistoricChoice%
CASE 4, 5:
  Text$ = Text$ + " pre-dating:"
  finish% = 13
CASE 8:
  Text$ = Text$ + " transforming:"
  finish% = 12
END SELECT
Present 1, "", Text$, 0, "C", 1, 1, 1
Text$ = "Press [ ESC ] to quit this menu."
Present 2, Text$, "", 0, "C", 1, 1, 25
Data$ = Load$(InputFile%, DataLength%, Choice&)
FOR c% = 1 TO finish%
  PresentDate Data$, c%, 1, HistoricChoice%
NEXT c%
current% = 1
Answer% = -1
DO
  DO
    LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
    Text$ = "Please choose a line by the arrow keys [" + CHR$(24) + "] or"
    Text$ = Text$ + " [" + CHR$(25) + "]"
    Present 2, Text$, "", 0, "C", 1, 1, 2
    Text$ = "and confirm your choice by the return key [ENTER]:"
    Present 2, Text$, "", 0, "C", 1, 1, 3
    PresentDate Data$, current%, 2, HistoricChoice%
    a$ = KeyInput$
    NextSt% = current%
    SELECT CASE ASC(a$)
    CASE 10, 13
      IF HistoricChoice% = 8 THEN
        Variable$ = "data.date[" + LTRIM$(STR$(INT((current% + 1) / 2)))
        IF current% MOD 2 = 1 THEN
          Variable$ = Variable$ + "].minimum"
        ELSE
          Variable$ = Variable$ + "].maximum"
        END IF
        IF Part$(Data$, Variable$) = SPACE$(SIZEOF$("term")) THEN
          a$ = CHR$(10)
        ELSE
          a$ = CHR$(13)
          Answer% = current%
        END IF
      ELSE
        a$ = CHR$(13)
        Answer% = current%
      END IF
    CASE 27
      a$ = CHR$(13)
      Answer% = 0
    END SELECT
    IF LEN(a$) = 2 THEN
```

```

a$ = RIGHT$(a$, 1)
SELECT CASE ASC(a$)
CASE 71      'Home
    NextSt% = 1
CASE 72
    IF current% > 1 THEN NextSt% = current% - 1 ELSE NextSt% = finish%
CASE 75
    NextSt% = INT((current% - 3) / 2) * 2 + 1
    IF NextSt% < 1 THEN NextSt% = finish%
CASE 77
    NextSt% = INT((current% + 1) / 2) * 2 + 1
    IF NextSt% > finish% THEN NextSt% = 1
CASE 79
    NextSt% = finish%
CASE 80
    IF current% < finish% THEN NextSt% = current% + 1 ELSE NextSt% = 1
CASE 83      'Del
    IF Question$("Delete record?", "N", InputSuccess%) = "Y" THEN
        IF current% = 13 AND InputSuccess% = 1 THEN
            Change Data$, "data.source", SPACE$(SIZEOF$("NAME"))
        ELSE
            IF InputSuccess% = 1 THEN
                current% = INT((current% + 1) / 2)
                Variable$ = "data.date[" + LTRIM$(STR$(current%)) + "]"
                Change Data$, Variable$, SPACE$(SIZEOF$("date"))
                current% = (current% - 1) * 2 + 1
                PresentDate Data$, current% + 1, 1, HistoricChoice%
            END IF
        END IF
    END IF
    Text$ = "Press [ ESC ] to quit this menu."
    Present 2, Text$, "", 0, "C", 1, 1, 25
END SELECT
PresentDate Data$, current%, 1, HistoricChoice%
current% = NextSt%
END IF
LOOP UNTIL a$ = CHR$(13)
LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
PresentDate Data$, current%, 1, HistoricChoice%
IF Answer% > 0 AND Answer% < 13 THEN
    Variable$ = "data.date[" + LTRIM$(STR$(INT((Answer% + 1) / 2)))
    Variable$ = Variable$ + "]"
    IF Answer% MOD 2 = 1 THEN
        DateValue$ = Part$(Data$, Variable$ + ".minimum")
    ELSE
        DateValue$ = Part$(Data$, Variable$ + ".maximum")
    END IF
    y$ = Part$(DateValue$, "term.sign") + Part$(DateValue$, "term.year")
    m$ = Part$(DateValue$, "term.month")
    d$ = Part$(DateValue$, "term.day")
    x% = CSRLIN
    ly% = POS(0) - SIZEOF$("term") - 1

    'Input of the Year Number:
    '-----'

    Text$ = "Input of the year number (unto four digits) with sign."
    Present 2, Text$, "", 0, "C", 1, 1, 2
    IF HistoricChoice% = 8 THEN

```

```

    Text$ = "0 is not valid."
ELSE
    Text$ = "0 means deleting."
END IF
Present 2, Text$, "", 0, "C", 1, 1, 3
repetition% = 1
DO
    LOCATE x%, ly%
    y$ = Editor$(y$, 5, 1, InputSuccess%)
    IF InputSuccess% = 1 THEN
        repetition% = 0
        IF VAL(y$) < 0 THEN
            y$ = LTRIM$(STR$(-VAL(y$)))
            y$ = "-" + SPACE$(4 - LEN(y$)) + y$
        END IF
    END IF
    IF VAL(y$) = 0 THEN
        IF HistoricChoice% = 8 THEN
            y$ = Part$(DateValue$, "term.sign")
            y$ = y$ + Part$(DateValue$, "term.year")
            repetition% = 1
            InputSuccess% = 1
        ELSE
            y$ = SPACE$(6)
            repetition% = 0
        END IF
    END IF
    IF VAL(y$) < -9999 OR VAL(y$) > 9999 THEN
        y$ = Part$(DateValue$, "term.sign") + Part$(DateValue$, "term.year")
        repetition% = 1
        InputSuccess% = 1
    END IF
LOOP UNTIL repetition% = 0 OR InputSuccess% = 0
IF VAL(y$) <> 0 THEN y$ = y$ + "."
COLOR 0, 7: PRINT y$;
LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
IF InputSuccess% = 0 THEN GOTO EndDatingLoop
IF VAL(y$) = 0 THEN
    m$ = SPACE$(3)
    d$ = SPACE$(3)
    GOTO ChangeDate
END IF

'Input of the Month Number:
'-----'
Text$ = "Input of the month number (between 1 and 12 inclusively).\"
Present 2, Text$, "", 0, "C", 1, 1, 2
IF HistoricChoice% = 8 THEN
    Text$ = "0 is not valid."
ELSE
    Text$ = "0 means deleting."
END IF
Present 2, Text$, "", 0, "C", 1, 1, 3
repetition% = 1
DO
    LOCATE x%, ly% + 6
    m$ = Editor$(m$, 2, 1, InputSuccess%)
    IF InputSuccess% = 1 THEN repetition% = 0

```

```
IF VAL(m$) = 0 THEN
  IF HistoricChoice% = 8 THEN
    m$ = Part$(DateValue$, "term.month")
    repetition% = 1
    InputSuccess% = 1
  ELSE
    m$ = SPACE$(3)
  END IF
END IF
IF VAL(m$) < 0 OR VAL(m$) > 12 THEN
  m$ = Part$(DateValue$, "term.month")
  repetition% = 1
  InputSuccess% = 1
END IF
LOOP UNTIL repetition% = 0 OR InputSuccess% = 0
IF VAL(m$) <> 0 THEN m$ = m$ + "."
COLOR 0, 7: PRINT m$;
LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
IF InputSuccess% = 0 THEN GOTO EndDatingLoop
IF VAL(m$) = 0 THEN
  y$ = SPACE$(6)
  d$ = SPACE$(3)
  GOTO ChangeDate
END IF

'Input of the Day Number:
'-----'
Text$ = "Input of the day number (between 1 and 30 inclusively).\"
Present 2, Text$, "", 0, "C", 1, 1, 2
IF HistoricChoice% = 8 THEN
  Text$ = "0 is not valid.\"
ELSE
  Text$ = "0 means deleting.\"
END IF
Present 2, Text$, "", 0, "C", 1, 1, 3
repetition% = 1
DO
  LOCATE x%, ly% + 9
  d$ = Editor$(d$, 2, 1, InputSuccess%)
  IF InputSuccess% = 1 THEN repetition% = 0
  IF VAL(d$) = 0 THEN
    IF HistoricChoice% = 8 THEN
      d$ = Part$(DateValue$, "term.day")
      repetition% = 1
      InputSuccess% = 1
    ELSE
      d$ = SPACE$(3)
    END IF
  END IF
  IF VAL(d$) < 0 OR VAL(d$) > 30 THEN
    d$ = Part$(DateValue$, "term.day")
    repetition% = 1
    InputSuccess% = 1
  END IF
LOOP UNTIL repetition% = 0 OR InputSuccess% = 0
IF VAL(d$) <> 0 THEN d$ = d$ + "."
COLOR 0, 7: PRINT d$;
LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
```

```

    IF InputSuccess% = 0 THEN GOTO EndDatingLoop
    IF VAL(d$) = 0 THEN
        y$ = SPACE$(6)
        m$ = SPACE$(3)
    END IF

'=====
ChangeDate:
'=====
    Term$ = y$ + m$ + d$
    IF Answer% MOD 2 = 1 THEN
        Change Data$, Variable$ + ".minimum", Term$
    ELSE
        Change Data$, Variable$ + ".maximum", Term$
    END IF
    minimum$ = Part$(Data$, Variable$ + ".minimum")
    maximum$ = Part$(Data$, Variable$ + ".maximum")
    Length% = SIZEOF%("term")
    IF minimum$ = SPACE$(Length%) AND maximum$ = SPACE$(Length%) THEN
        Status$ = " "
    ELSEIF HistoricChoice% <> 8 THEN
        IF minimum$ = SPACE$(Length%) THEN
            Status$ = CHR$(222) 'block rhs
        ELSEIF maximum$ = SPACE$(Length%) THEN
            Status$ = CHR$(221) 'block lhs
        ELSE
            Status$ = CHR$(219) 'full block
        END IF
    END IF
    Change Data$, Variable$ + ".status", Status$
ELSEIF Answer% = 13 THEN

    'Add Source Entry:
    '-----'
    Length% = SIZEOF%("NAME")
    x% = CSRLIN
    ly% = POS(0) - Length% - 1
    source$ = Part$(Data$, "data.source")
    Text$ = "Please add the source of your pre-dating"
    Present 2, Text$, "", 0, "C", 1, 1, 2
    Text$ = "to the 1st source entry as far as possible:"
    Present 2, Text$, "", 0, "C", 1, 1, 3
    repetition% = 1
    DO
        LOCATE x%, ly%
        source$ = Editor$(source$, Length%, 3, InputSuccess%)
    LOOP UNTIL source$ <> SPACE$(Length%) OR InputSuccess% = 0
    IF InputSuccess% = 1 THEN
        Change Data$, "data.source", source$
    END IF
    LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
END IF

'=====
EndDatingLoop:
'=====

'Check, whether the 1st Source must be Deleted:
'-----'

```

```

AddSource% = 0
FOR c% = 1 TO 6
    Variable$ = "data.date[" + LTRIM$(STR$(c%)) + "].status"
    Status$ = Part$(Data$, Variable$)
    SELECT CASE Status$
        CASE CHR$(219), CHR$(221), CHR$(222)      'several blocks
            AddSource% = 1
            c% = 6
        END SELECT
    NEXT c%
    IF AddSource% = 0 AND HistoricChoice% <> 8 THEN

        'No Predecessor mentioned:
        '-----'
        IF Part$(Data$, "data.p[1].direct") = " " THEN

            'Check, whether the Character " = " Occurs within the Name Entry:
            '-----'
            IF INSTR(Part$(Data$, "data.name"), " = ") > 0 THEN
                AddSource% = 1
            END IF
        END IF
    END IF
    IF AddSource% = 0 THEN
        Change Data$, "data.source", SPACE$(SIZEOF$("NAME"))
        PresentDate Data$, 13, 1, HistoricChoice%
    END IF

    LOOP UNTIL Answer% = 0 OR HistoricChoice% = 8
    IF HistoricChoice% <> 8 THEN
        WriteTo InputFile%, DataLength%, Choice%, Data$
    END IF

    Dating% = Answer%
END FUNCTION 'Dating% _____'

'=====
FUNCTION Editor$ (Default$, Length%, Mode%, InputSuccess%)
'=====
' `Mode%` = 1 means number input
' `Mode%` = 2 means text input
' `Mode%` = 3 means character input
' `InputSuccess%` = 0 means termination by [ESC],
' `InputSuccess%` = 1 means successful input
'
' Handling:
' 9/ 4/2001 - 3/27/2003: Norbert Südland
' Translation:
' 10/ 9/2007: Norbert Südland
'-----'

DIM Answer$ 'AS STRING
DIM a$
DIM p% 'AS INTEGER
DIM l%
DIM x%
DIM ly%
DIM start%
DIM insert%

```

```

DIM digit% 'AS LONG

InputSuccess% = 1

'Current Position:
'-----'

p% = 1
IF Mode% > 1 THEN
    Default$ = LTRIM$(Default$)
    Default$ = RTRIM$(Default$)
END IF
l% = LEN(Default$)
IF l% < Length% THEN
    Default$ = Default$ + SPACE$(Length% - l%)
ELSE
    Default$ = LEFT$(Default$, Length%)
    l% = Length%
END IF
Answer$ = Default$
COLOR 0, 7
x% = CSRLIN
ly% = POS(0)
LOCATE x%, ly%, 1, 12, 13
start% = 1
insert% = 1
DO
    LOCATE x%, ly%
    PRINT Answer$;
    LOCATE x%, ly% + p% - 1, 1
    a$ = KeyInput$
    IF LEN(a$) = 1 THEN
        SELECT CASE a$
            CASE CHR$(0), CHR$(3) TO CHR$(7), CHR$(9), CHR$(11), CHR$(12)
                GOTO EndofLoop
                'unused characters
            CASE CHR$(26), CHR$(28) TO CHR$(31), CHR$(255)
                GOTO EndofLoop
                'unused characters
            CASE CHR$(8)
                IF p% > 1 THEN
                    p% = p% - 1
                    IF l% > 1 THEN
                        l% = l% - 1
                    END IF
                    MID$(Answer$, p%, Length% - p%) = RIGHT$(Answer$, Length% - p%)
                    MID$(Answer$, Length%, 1) = " "
                END IF
            CASE CHR$(10), CHR$(13)
                a$ = CHR$(13)
                InputSuccess% = 1
                GOTO EndofLoop
            CASE CHR$(27)
                'put back mask or quit without damage
                IF Mode% = 1 AND LEN(LTRIM$(Answer$)) = 0 AND p% = 1 THEN
                    InputSuccess% = 0
                    GOTO EndofLoop
                END IF
        END SELECT
    END IF

```

```
IF Answer$ = Default$ AND l% = LEN(RTRIM$(Default$)) AND p% = 1 THEN
    InputSuccess% = 0
    GOTO EndofLoop
END IF
Answer$ = Default$
l% = LEN(RTRIM$(Default$))
p% = 1
CASE "-", "+"
    IF Mode% = 1 THEN
        IF Length% > 1 THEN
            IF p% = 1 GOTO SaveCharacter
        ELSE
            IF Length% = 1 THEN
                IF a$ = "-" THEN
                    Answer$ = LTRIM$(STR$( (VAL(Answer$) + 9) MOD 11))
                ELSE
                    Answer$ = LTRIM$(STR$( (VAL(Answer$) + 1) MOD 11))
                END IF
            END IF
            InputSuccess% = 0
        END IF
    END IF
ELSE
    IF Mode% = 3 THEN
        GOTO SaveCharacter
    END IF
END IF
CASE "0" TO "9"
    GOTO SaveCharacter
CASE " " TO ")", "@", "Z", "_", "a" TO "z", "ç" TO CHR$(254)
    IF Mode% > 1 THEN
        GOTO SaveCharacter
    END IF
CASE CHR$(1) TO CHR$(254)
    IF Mode% = 3 THEN
        GOTO SaveCharacter
    END IF
END SELECT
END IF
IF LEN(a$) = 2 THEN
    'Control Keys
    a$ = RIGHT$(a$, 1)
    SELECT CASE ASC(a$)
    CASE 59 TO 67
        '[ F1 ] to [ F9 ]
        IF Mode% = 1 THEN
            a$ = CHR$(ASC("1") + ASC(a$) - 59)
            GOTO SaveCharacter
        END IF
    CASE 71
        'Position Home
        IF Length% > 1 THEN
            p% = 1
            start% = 0
        ELSE
            IF Mode% = 1 AND Length% = 1 THEN
                Answer$ = LTRIM$(STR$(1))
                InputSuccess% = 0
            END IF
        END IF
    END IF
```

```
END IF
CASE 72
  'Cursor Up
  IF Mode% = 1 AND Length% = 1 THEN
    Answer$ = LTRIM$(STR$(VAL(Answer$) + 9) MOD 10))
    InputSuccess% = 0
  END IF
CASE 75
  'Cursor Left
  IF Length% > 1 THEN
    IF p% > 1 THEN
      p% = p% - 1
    END IF
    start% = 0
  ELSE
    IF Mode% = 1 AND Length% = 1 THEN
      Answer$ = LTRIM$(STR$(VAL(Answer$) + 9) MOD 10))
      InputSuccess% = 0
    END IF
  END IF
CASE 77
  'Cursor Right
  IF Length% > 1 THEN
    IF p% < Length% AND p% <= 1% THEN
      p% = p% + 1
    END IF
    start% = 0
  ELSE
    IF Mode% = 1 AND Length% = 1 THEN
      Answer$ = LTRIM$(STR$(VAL(Answer$) + 1) MOD 11))
      InputSuccess% = 0
    END IF
  END IF
CASE 79
  'Position End
  IF Length% > 1 THEN
    IF l% < Length% THEN
      p% = l% + 1
    ELSE
      p% = l%
    END IF
    start% = 0
  ELSE
    IF Mode% = 1 AND Length% = 1 THEN
      Answer$ = LTRIM$(STR$(0))
      InputSuccess% = 0
    END IF
  END IF
CASE 80
  'Cursor Down
  IF Mode% = 1 AND Length% = 1 THEN
    Answer$ = LTRIM$(STR$(VAL(Answer$) + 1) MOD 11))
    InputSuccess% = 0
  END IF
CASE 82
  'Insert
  IF Length% > 1 THEN
    start% = 0
```

```
        IF insert% = 1 THEN
            insert% = 0
            LOCATE x%, ly% + p% - 1, 1, 0, 13
        ELSE
            insert% = 1
            LOCATE x%, ly% + p% - 1, 1, 12, 13
        END IF
    END IF
CASE 83
    'Del(ete)
    IF Length% > 1 THEN
        start% = 0
        MID$(Answer$, p%, Length% - p%) = RIGHT$(Answer$, Length% - p%)
        MID$(Answer$, Length%, 1) = " "
        IF l% > 1 THEN
            l% = l% - 1
        END IF
    END IF
END SELECT
END IF
GOTO EndofLoop

'=====
SaveCharacter:
'=====
    IF Length% = 1 THEN
        InputSuccess% = 1
        Answer$ = a$
        a$ = CHR$(13)
    ELSE
        IF start% = 1 THEN
            start% = 0
            Answer$ = SPACE$(Length%)
            l% = 0
        END IF
        IF insert% = 1 THEN
            IF l% <= Length% THEN
                IF l% < Length% THEN
                    l% = l% + 1
                END IF
                MID$(Answer$, p%, l% - p% + 1) = a$ + MID$(Answer$, p%, l% - p%)
            END IF
        ELSE
            MID$(Answer$, p%, 1) = a$
            IF p% = l% AND l% < Length% THEN
                l% = l% + 1
            END IF
        END IF
        IF p% < Length% THEN
            p% = p% + 1
        END IF
    END IF

'=====
EndofLoop:
'=====
    LOOP UNTIL a$ = CHR$(13) OR InputSuccess% = 0

    IF InputSuccess% = 1 THEN
```

```

IF Mode% THEN
  IF Mode% = 1 THEN
    digit& = VAL(Answer$)
    Answer$ = LTRIM$(STR$(digit&))
    Answer$ = SPACE$(Length% - LEN(Answer$)) + Answer$
  END IF
  IF Mode% = 2 + 3 THEN
    Answer$ = LTRIM$(Answer$)
    Answer$ = Answer$ + SPACE$(Length% - LEN(Answer$))
  END IF
END IF
END IF
END IF
COLOR 7, 0
LOCATE x%, ly%, 0

Editor$ = Answer$
END FUNCTION 'Editor$ _____'

'=====
FUNCTION FillUpText$ (Text$, WantedLength%, Character$)
'=====
' Will fill up the `Text$` by `Character$` until `WantedLength%` is
' reached, where a positive `WantedLength%` will fill up to the left
' hand side, and a negative `WantedLength%` to the right hand side.
'
' If `Character$` is a STRING of length zero, then the length of `Text$`
' sets the maximum of `WantedLength`.
'
' So far problems with memory overflow for extremely long character chains
' are not cached, because they will depend on the configuration of the
' calling BASIC interpreter!
'
'
' Handling:
' 8/18/2001: Norbert Südland, Adelshofen
' Check:
' 8/18/2001: Norbert Südland, Adelshofen
' Translation:
' 10/ 9/2007: Norbert Südland, Aalen
'-----
DIM result$          'AS STRING

'Preparation:
'-----
result$ = Text$

'Fill Up the Text:
'-----
IF WantedLength% < 0 THEN

  'To the Right Hand Side:
  '-----
  IF LEN(Character$) > 0 THEN
    WHILE LEN(result$) < -WantedLength%
      result$ = Character$ + result$
    WEND
  END IF

```

```

    result$ = RIGHT$(result$, -WantedLength%)
ELSE
    'To the Left Hand Side:
    '-----'
    IF LEN(Character$) > 0 THEN
        WHILE LEN(result$) < WantedLength%
            result$ = result$ + Character$
        WEND
    END IF
    result$ = LEFT$(result$, WantedLength%)
END IF

'Result:
'-----'
FillUpText$ = result$
END FUNCTION 'FillUpText$ _____ '

'=====
FUNCTION KeyInput$
'=====
' Waits for a keyboard input and gives back the corresponding ASCII letter.
'
' Handling:
' 8/18/2001: Norbert Südland and Eckhard Walter, Adelshofen
' Check:
' 8/18/2001: Norbert Südland and Eckhard Walter, Adelshofen
' 9/ 6/2002: Norbert Südland, Aalen
'-----'
DIM Answer$          'AS STRING

'Empty the keyboard buffer:
'-----'
WHILE INKEY$ <> ""
WEND

'Question keyboard again, until a key has been pressed:
'-----'
Answer$ = ""
WHILE LEN(Answer$) = 0
    Answer$ = INKEY$
WEND

'Result:
'-----'
KeyInput$ = Answer$
END FUNCTION 'KeyInput$ _____ '

'=====
SUB LINEINPUT (File%, DataLine$)
'=====
' Replaces the command LINE INPUT to run also under the DOS Box 0.73.
' This box generates by DIR an output like under Unix and can be evaluated
' by LINEINPUT instead of LINE INPUT.
' Therefore the 'File%' must be opened FOR ACCESS BINARY READ (WRITE).
' At the end a line of 512 characters is returned, thus each line not
' ending by CHR$(10) is taken. This compromise enables speed and is

```

```

' sufficient here to evaluate the DIR command.
'
' Handling:
' 8/11/2009:          Norbert Südland, D-73431 Aalen
' Check:
' 8/13/2009:          Norbert Südland, D-73431 Aalen
' Translation:
' 8/11/2009 - 8/13/2009: Norbert Südland, D-73431 Aalen
'-----'
DIM OldPosition&
DIM Length&
DIM NewPosition&

DataLine$ = SPACE$(512)
OldPosition& = SEEK(File%)
GET #File%, OldPosition&, DataLine$
Length& = INSTR(DataLine$, CHR$(10))
IF Length& <> 0 THEN
    DataLine$ = LEFT$(DataLine$, Length& - 1)
    IF INSTR(RIGHT$(DataLine$, 1), CHR$(13)) <> 0 THEN
        DataLine$ = LEFT$(DataLine$, LEN(DataLine$) - 1)
    END IF
END IF

IF EOF(File%) = 0 OR Length& <> 0 THEN
    NewPosition& = OldPosition& + Length&
    SEEK #File%, NewPosition&
END IF
'PRINT LEN(DataLine$), DataLine$
'STOP
END SUB 'LINEINPUT _____'

'=====
FUNCTION Load$ (File%, DataSetLength%, Position&)
'=====
' Will load the `Position`th record of a binary `File%`, whose
' `DataSetLength%` is constant.
' The 0th record is used for documentation.
'
' Handling:
' 1/21/2003: Norbert Südland
' Translation:
' 10/ 9/2007: Norbert Südland
'-----'
DIM Data$ 'AS STRING

Data$ = SPACE$(DataSetLength%) 'Generate space!
GET #File%, Position& * DataSetLength% + 1, Data$

Load$ = Data$
END FUNCTION 'Load$ _____'

'=====
FUNCTION Moment% (symbol$)
'=====
' Will transform the moment symbol into an ordinary number of date.
' This function will work correctly with old data files, too.
'

```

```

' Handling:
' 8/ 5/2001 - 9/ 4/2001: Norbert Südland, Munich
' Translation:
' 10/ 9/2007 - 11/19/2007: Norbert Südland, Aalen
' Addition:
' 11/20/2007: Norbert Südland
'-----
DIM MomentNumber%

SELECT CASE symbol$
CASE "{" , "*"
    MomentNumber% = 1
CASE "[" , "A" , "B"
    MomentNumber% = 2
CASE "(" , "W" , CHR$(224) 'alpha
    MomentNumber% = 3
CASE ")" , "X" , CHR$(234) 'Omega
    MomentNumber% = 4
CASE "]" , "E"
    MomentNumber% = 5
CASE "}" , "+"
    MomentNumber% = 6
CASE ELSE
    MomentNumber% = 0
END SELECT
Moment% = MomentNumber%
END FUNCTION 'Moment%

'=====
FUNCTION NameEntry$ (Position&)
'=====
' Will read the `Position&`th name entry from the `InputFile%`.
' The 0th file record is used for documentation.
'
' Handling:
' 12/17/2002 - 1/21/2003: Norbert Südland
' Translation:
' 10/ 9/2007: Norbert Südland
'-----
DIM Data$ 'AS STRING

Data$ = SPACE$(DataLength%)
701 GET #InputFile%, Position& * DataLength% + 1, Data$
NameEntry$ = Part$(Data$, "data.name")
END FUNCTION 'NameEntry$

'=====
SUB NewStatus (DateValue$)
'=====
'
' Handling:
' 9/ 4/2001: Norbert Südland, Munich
' Translation:
' 10/ 9/2007: Norbert Südland, Aalen
'-----
DIM Status$ 'AS STRING

```

```

DIM Length%          'AS INTEGER

SELECT CASE Part$(DateValue$, "date.status")
CASE CHR$(219), CHR$(221), CHR$(222) 'no change of pre-dated entries!
CASE ELSE
    Length% = SIZEOF$("term")
    IF Part$(DateValue$, "date.minimum") = SPACE$(Length%) THEN
        Status$ = CHR$(243)          '<=
    ELSE
        Status$ = SPACE$(1)
    END IF
    IF Part$(DateValue$, "date.maximum") = SPACE$(Length%) THEN
        IF Status$ = " " THEN
            Status$ = CHR$(242)      '>=
        ELSE
            Status$ = SPACE$(1)
        END IF
    END IF
    Change DateValue$, "date.status", Status$
END SELECT
END SUB 'NewStatus _____

'=====
'FUNCTION Number% (Text$, start%, finish%)
'=====
' Will seek for the data struct `Text$` and will give back the location.
' The list `GVName$()` must be sorted.
'
' Handling:
' 8/ 5/2001 - 9/ 4/2001:    Norbert Südland, Munich
' 12/18/2002:              Norbert Südland, Aalen
' Translation:
' 10/ 9/2007:              Norbert Südland, Aalen
'-----

DIM Found%          'AS INTEGER
DIM begin%
DIM endpos%
DIM midpos%

'Preparation:
'-----
Found% = 0
begin% = start%
endpos% = finish%

'Construct the New Search Area:
'-----
WHILE Found% = 0 AND begin% <= endpos%
    midpos% = INT(begin% / 2 + endpos% / 2)
    SELECT CASE GVName$(midpos%)
    CASE IS = Text$
        Found% = midpos%
    CASE IS < Text$
        begin% = midpos% + 1
    CASE IS > Text$
        endpos% = midpos% - 1
    END SELECT
END WHILE

```

```

    END SELECT
WEND

'Result:
'-----'
Number% = Found%
END FUNCTION 'Number% _____'

'=====
FUNCTION NumberToTerm$ (Sign%, Year%, Month%, Day%)
'=====
' Will convert a row of date numbers to a string.
'
' Handling:
' 9/ 4/2001: Norbert Südland, Munich
' Translation:
' 10/ 9/2007: Norbert Südland, Aalen
'-----
DIM Term$      'AS STRING

Term$ = SPACE$(SIZEOF$("term"))
Year% = Sign% * Year%
IF Year% < 0 THEN
    Sign% = -1
    Year% = -Year%
ELSE
    Sign% = 1
END IF
IF Sign% = 1 THEN
    Change Term$, "term.sign", " "
ELSE
    Change Term$, "term.sign", "-"
END IF
IF Year% <> 0 THEN
    Change Term$, "term.year", LTRIM$(STR$(Year%))
    Change Term$, "term.ys", "."
END IF
IF Month% <> 0 THEN
    Change Term$, "term.month", LTRIM$(STR$(Month%))
    Change Term$, "term.ms", "."
END IF
IF Day% <> 0 THEN
    Change Term$, "term.day", LTRIM$(STR$(Day%))
    Change Term$, "term.ds", "."
END IF
NumberToTerm$ = Term$
END FUNCTION 'NumberToTerm$ _____'

'=====
FUNCTION OrdinaryNumber$ (Num%)
'=====
' Will generate the correct ending of a written ordinary number.
'
' Handling:
' 9/15/2007: Norbert Südland
' Check:
' 9/15/2007: Norbert Südland

```

```

'-----'
DIM current&
DIM result$

result$ = LTRIM$(STR$(Num%))
IF Num% < 0 THEN
    current& = Num%
    current& = (-current&) MOD 100
ELSE
    current& = Num% MOD 100
END IF
IF current& - (current& MOD 10) <> 10 THEN
    SELECT CASE current& MOD 10
    CASE 1
        result$ = result$ + "st"
    CASE 2
        result$ = result$ + "nd"
    CASE 3
        result$ = result$ + "rd"
    CASE ELSE
        result$ = result$ + "th"
    END SELECT
ELSE
    result$ = result$ + "th"
END IF

OrdinaryNumber$ = result$
END FUNCTION 'OrdinaryNumber$-----'

'=====
FUNCTION Overview& (HistoricChoice%, Default&, InputSuccess%)
'=====
' Will present an overview on the records of a file.
'
' Handling:
' 8/ 5/2001 - 2/10/2003:    Norbert Südland
' Check:
' 12/17/2002:              Norbert Südland, Aalen
' Translation:
' 10/18/2007:              Norbert Südland
'-----'

DIM Text1$           'AS STRING
DIM Text2$
DIM Name$
DIM a$
DIM c%               'AS INTEGER
DIM Answer&          'AS LONG INTEGER
DIM NextSt&
DIM start&
DIM endpos&
DIM current&

' Exclude an Empty File:
'-----'
IF GIL& <= 0 THEN
    Answer& = 0
    InputSuccess% = 0

```

```

GOTO EndofOverview
ELSE
    current& = FNMIN(FNMAX(1&, Default&), GIL&)

    ' Hint: `CLNG( )` Will Work Strangely with QuickBasic 4.5:
    '-----'
    start& = current& - ((current& - 1) MOD 57)
    endpos& = FNMIN(GIL&, current& - ((current& - 1) MOD 57) + 59&)
    IF GIL& > 60 AND GIL& - start& < 59 THEN
        start& = GIL& - 59
        endpos& = GIL&
    END IF
END IF

DO
    ' Menu Presentation:
    '-----'
    CLS
    SELECT CASE HistoricChoice%
    CASE 3
        Text1$ = "(3)...Delete a Data Record"
        Text2$ = "Please Choose a Name:"
    CASE 4
        Text1$ = "(4)...Pre-Date a Data Record"
        Text2$ = "Please Choose a Name:"
    CASE 5
        Text1$ = "(5)...Overview and Correction"
        Text2$ = "Please Choose a Name:"
    CASE 6
        Text1$ = "(7)...Sort the Data File"
        Text2$ = "Please Choose a Name:"
    CASE 7
        Text1$ = "(7)...Sort the Data File"
        Text2$ = "Please Choose a Name:"
    CASE 8
        Text1$ = "(8)...Transform the Result"
        Text2$ = "Please Choose a Date for Explicitly Transforming:"
    END SELECT
    Present 1, "", Text1$, 0, "C", 1, 1, 1
    Present 2, Text2$, "", 0, "C", 1, 1, 2
    Text1$ = "Please press [ ESC ] to quit the menu."
    Present 2, Text1$, "", 0, "C", 1, 1, 25

    ' Presentation Window:
    '-----'
    FOR c% = 0 TO endpos& - start&
        Name$ = NameEntry$(start& + c%)
        Present 1, Name$, "", 0, "C", (c% MOD 3) + 1, 3, INT(c% / 3) + 4
    NEXT c%

    ' Evaluate the Key Input:
    '-----'
    Answer& = 0
    DO
        Name$ = " " + NameEntry$(current&) + " "
        c% = INT(current& - start&)
        Present 2, "", Name$, 0, "C", c% MOD 3 + 1, 3, INT(c% / 3) + 4
        a$ = KeyInput$

```

```
NextSt& = current&
SELECT CASE ASC(a$)
CASE 10, 13
    a$ = CHR$(13)
    Answer& = current&
    InputSuccess% = 1
CASE 27
    a$ = CHR$(13)
    Answer& = current&
    InputSuccess% = 0
END SELECT
IF LEN(a$) = 2 THEN
    a$ = RIGHT$(a$, 1)
    SELECT CASE ASC(a$)
    CASE 71 'Home
        NextSt& = start&
    CASE 72 'Cursor up
        IF current& - 3 >= start& THEN
            NextSt& = current& - 3
        ELSEIF GIL& > 60 THEN
            a$ = CHR$(13)
            NextSt& = FNMAX(1&, current& - 3)
            start& = start& - 3
            endpos& = endpos& - 3
            IF start& < 1 THEN
                NextSt& = current&
                start& = 1
                endpos& = 60
            END IF
        END IF
    CASE 73 'Page up
        IF start& = 1 THEN
            NextSt& = 1
        ELSEIF GIL& > 60 THEN
            a$ = CHR$(13)
            NextSt& = FNMAX(1&, current& - 57)
            start& = start& - 57
            endpos& = endpos& - 57
            IF start& < 1 THEN
                start& = 1
                endpos& = 60
            END IF
        END IF
    CASE 75 'Cursor left
        IF current& > start& THEN
            NextSt& = current& - 1
        ELSEIF GIL& > 60 THEN
            a$ = CHR$(13)
            NextSt& = FNMAX(1&, current& - 1)
            start& = start& - 3
            endpos& = endpos& - 3
            IF start& < 1 THEN
                start& = 1
                endpos& = 60
            END IF
        END IF
    CASE 77 'Cursor right
        IF current& < endpos& THEN
```

```
    NextSt& = current& + 1
ELSEIF GIL& > 60 THEN
    a$ = CHR$(13)
    NextSt& = FNMIN(current& + 1, GIL&)
    start& = start& + 3
    endpos& = endpos& + 3
    IF endpos& > GIL& THEN
        start& = GIL& - 59
        endpos& = GIL&
    END IF
END IF
CASE 79
    NextSt& = endpos&
CASE 80    'Cursor down
    IF current& + 3 <= endpos& THEN
        NextSt& = current& + 3
    ELSEIF GIL& > 60 THEN
        a$ = CHR$(13)
        NextSt& = FNMIN(current& + 3, GIL&)
        start& = start& + 3
        endpos& = endpos& + 3
        IF endpos& > GIL& THEN
            NextSt& = current&
            start& = GIL& - 59
            endpos& = GIL&
        END IF
    END IF
CASE 81    'Page down
    IF endpos& = GIL& THEN
        NextSt& = endpos&
    ELSEIF GIL& > 60 THEN
        a$ = CHR$(13)
        NextSt& = FNMIN(current& + 57, GIL&)
        start& = start& + 57
        endpos& = endpos& + 57
        IF endpos& > GIL& THEN
            start& = GIL& - 59
            endpos& = GIL&
        END IF
    END IF
CASE 117, 118    '[ Ctrl ] - [ End ], [ Ctrl ] - [ Page down ]
    IF endpos& = GIL& THEN
        NextSt& = GIL&
    ELSEIF GIL& > 60 THEN
        a$ = CHR$(13)
        NextSt& = GIL&
        start& = GIL& - 59
        endpos& = GIL&
    END IF
CASE 119, 132    '[ Strg ] - [ Home ], [ Strg ] - [ Page up ]
    IF start& = 1 THEN
        NextSt& = 1
    ELSEIF GIL& > 60 THEN
        a$ = CHR$(13)
        NextSt& = 1
        start& = 1
        endpos& = 60&
    END IF
```

```

        END SELECT
        Present 1, Name$, "", 0, "C", c% MOD 3 + 1, 3, INT(c% / 3) + 4
        Name$ = NameEntry$(NextSt&)
        current& = NextSt&
    END IF
    LOOP UNTIL a$ = CHR$(13)
    LOOP UNTIL Answer& > 0

'=====
EndofOverview:
'=====
    Overview& = Answer&
END FUNCTION 'Overview& _____

'=====
FUNCTION Part$ (Data$, Variable$)
'=====
' Will extract the part of `Data$` that is given by `Variable$`.
'
' Handling:
' 9/ 4/2001: Norbert Südland, Munich
' 12/18/2002: Norbert Südland, Aalen
' Translation:
' 11/ 9/2007: Norbert Südland
'-----
DIM seek$           'AS STRING
DIM result$
DIM FurtherSeek$
DIM Length%         'AS INTEGER
DIM p%
DIM finish%
DIM where%

result$ = Data$
FurtherSeek$ = Variable$
DO
    Length% = LEN(FurtherSeek$)
    p% = STRLEN%(FurtherSeek$, ".")
    IF p% = Length% THEN
        seek$ = FurtherSeek$
    ELSE
        p% = p% + 1
        p% = p% + STRLEN%(MID$(FurtherSeek$, p% + 1, Length% - p%), ".")
        IF p% < Length% THEN
            seek$ = LEFT$(FurtherSeek$, p%)
        ELSE
            seek$ = FurtherSeek$
        END IF
    END IF
    where% = Number%(seek$, 1, GlobVarNumber%)
    IF where% = 0 THEN
        result$ = ""
    ELSE
        result$ = MID$(result$, GVBEGIN%(where%), GVLENGTH%(where%))
        IF p% < Length% THEN
            FurtherSeek$ = MID$(FurtherSeek$, p% + 1, Length% - p%)
            FurtherSeek$ = GVTYPE$(where%) + FurtherSeek$
        END IF
    END IF

```

```

        END IF
    END IF
    LOOP UNTIL p% = Length% OR where% = 0

    Part$ = result$
END FUNCTION 'Part$ _____'

'=====
SUB Pause
'=====
' Will present a statement in line 25 and wait for a pressed key.
'
' Handling:
'   8/18/2001 Norbert Südland and Eckhard Walter, Adelshofen
' Check:
'   8/18/2001 Norbert Südland and Eckhard Walter, Adelshofen
' Translation:
' 11/ 9/2007 Norbert Südland
'-----
DIM x%      'AS INTEGER
DIM y%      'AS INTEGER
DIM Answer$ 'AS STRING

'Find Current Cursor Position:
'-----
x% = CSRLIN
y% = POS(0)

'Delete Line 25 and Present the Statement:
'-----
LOCATE 25, 1, 0
PRINT SPACE$(80);
Present 1, "", "Press any key to go on", 0, "C", 1, 1, 25

'Wait for a Pressed Key:
'-----
Answer$ = KeyInput$

'Delete Line 25 Again:
'-----
LOCATE 25, 1, 1
PRINT SPACE$(80);

'Restore Cursor Position:
'-----
LOCATE x%, y%
END SUB 'Pause _____'

'=====
SUB Present (Colors%, Quest$, Buffer$, Offset%, Kind$, Area%, Areas%, ly%)
'=====
' Presents `Quest$`, followed by `Buffer$` in line `ly%`.
'
' Meaning of the further parameters:
' `Colors%`   Choice mode 1 (normal) or 2 (emphasized)
' `Offset%`   Shifting possibility within a column
' `Area%`     Wanted column

```

```

' `Areas%`      Total number of columns
' `Kind$`       "L" (left hand sided), "C" (centered), "R" (right hand sided)
'
' Handling:
' 8/ 7/2001 - 9/ 4/2001: Norbert Südland and Eckhard Walter, Adelshofen
'-----'
DIM Length%    'AS INTEGER
DIM x%

SELECT CASE Colors%
CASE 1
  COLOR 7, 0
CASE 2
  COLOR 15, 0
END SELECT
Length% = LEN(Quest$) + LEN(Buffer$) + 2
IF LEN(Buffer$) = 0 THEN
  Length% = Length% - 2
END IF
IF Length% > 80 / Areas% THEN
  Length% = 80 / Areas%
  IF LEN(Quest$) >= Length% THEN      'Programming Discrepancy!
    Quest$ = LEFT$(Quest$, Length%)
    Buffer$ = ""
  ELSE
    Pause
    Buffer$ = LEFT$(Buffer$, LEN(Quest$) - Length%)
  END IF
END IF

SELECT CASE Kind$
CASE "L"
  x% = INT(Offset% + (Area% - 1) * 80 / Areas%)
CASE "C"
  x% = INT(Offset% + (Area% - .5) * 80 / Areas% - Length% / 2) + 1
CASE "R"
  x% = INT(Offset% + Area% * 80 / Areas% - Length%)
END SELECT
LOCATE ly%, x%
PRINT Quest$;

IF Buffer$ <> "" THEN
  COLOR 0, 7
  SELECT CASE Colors%
CASE 1
  PRINT " "; Buffer$; " ";
CASE 2
  IF Buffer$ <> "" THEN
    x% = POS(0)
    LOCATE ly%, x% + 1
    PRINT Buffer$;
  END IF
END SELECT
END IF

COLOR 7, 0
END SUB 'Present _____'

```

```

'=====
SUB PresentDate (Data$, c%, Colors%, HistoricChoice%)
'=====
'
' Handling:
' 9/ 4/2001 - 12/16/2002: Norbert Suedland
' 4/18/2007: Norbert Suedland
' Translation:
' 11/ 9/2007: Norbert Suedland
'-----
DIM Variable$ 'AS STRING
DIM DateValue$
DIM Text$
DIM Status$
DIM LineNumber%
DIM show%

IF c% < 13 THEN
    Variable$ = "data.date[" + LTRIM$(STR$(INT((c% + 1) / 2))) + "]"
    Status$ = Part$(Data$, Variable$ + ".status")
    IF HistoricChoice% = 8 OR Status$ = CHR$(219) THEN 'full block
        show% = 1
    ELSE
        show% = 0
        IF Status$ = CHR$(221) AND c% MOD 2 = 1 THEN 'block lhs
            show% = 1
        END IF
        IF Status$ = CHR$(222) AND c% MOD 2 = 0 THEN 'block rhs
            show% = 1
        END IF
    END IF
    IF show% = 1 THEN
        SELECT CASE (c% - 1) MOD 2 + 1
        CASE 1
            Variable$ = Variable$ + ".minimum"
        CASE 2
            Variable$ = Variable$ + ".maximum"
        CASE ELSE
            STOP 'programming mistake
        END SELECT
        DateValue$ = Part$(Data$, Variable$)
    ELSE
        DateValue$ = SPACE$(SIZEOF%("term"))
    END IF
ELSE
    Variable$ = "data.source"
    DateValue$ = Part$(Data$, Variable$)
END IF
SELECT CASE c%
CASE 1
    Text$ = "Begin of Life (*) Minimum: "
CASE 2
    Text$ = "Begin of Life (*) Maximum: "
CASE 3
    Text$ = "Begin of 1st Aera (B) Minimum: "
CASE 4

```

```

    Text$ = "Begin of 1st Aera  (B) Maximum: "
CASE 5
    Text$ = "Begin of 2nd Aera  (" + CHR$(224) + ") Minimum: "
CASE 6
    Text$ = "Begin of 2nd Aera  (" + CHR$(224) + ") Maximum: "
CASE 7
    Text$ = "End of 2nd Aera    (" + CHR$(234) + ") Minimum: "
CASE 8
    Text$ = "End of 2nd Aera    (" + CHR$(234) + ") Maximum: "
CASE 9
    Text$ = "End of 1st Aera    (E) Minimum: "
CASE 10
    Text$ = "End of 1st Aera    (E) Maximum: "
CASE 11
    Text$ = "End of Life        (+) Minimum: "
CASE 12
    Text$ = "End of Life        (+) Maximum: "
CASE 13
    Text$ = "Source: "
END SELECT
LineNumber% = INT((c% - 1) / 2) * 3 + 4 + (c% - 1) MOD 2 + 1
Present Colors%, Text$, DateValue$, 0, "C", 1, 1, LineNumber%
END SUB 'PresentDate _____

'=====
SUB PresentLine (Data$, c%, Colors%)
'=====
' Will present a line of the file record overview.
'
' Handling:
' 9/ 4/2001: Norbert Südland, Munich
' Translation:
' 11/ 9/2007 - 1/24/2009: Norbert Südland
'-----

DIM Quest$      'AS STRING
DIM Variable$
DIM Buffer$
DIM Offset%     'AS INTEGER
DIM Area%
DIM Areas%
DIM ly%

Offset% = 0
SELECT CASE c%
CASE 1
    Quest$ = "Name /Event: "
    Variable$ = "data.name"
    Area% = 1
    Areas% = 1
    ly% = 5
CASE 2, 6
    Area% = (c% - 2) / 4 + 1
    Areas% = 2
    Quest$ = "Moment for " + OrdinaryNumber$(Area%) + " Predecessor: "
    Variable$ = "data.p[" + LTRIM$(STR$(Area%)) + "].moment1"
    ly% = 6
CASE 3, 7

```

```
Area% = (c% - 3) / 4 + 1
Areas% = 2
Quest$ = "          Directly Before? [Y/N] "
Variable$ = "data.p[" + LTRIM$(STR$(Area%)) + "].direct"
ly% = 7
CASE 4, 8
Area% = (c% - 4) / 4 + 1
Areas% = 2
Quest$ = OrdinaryNumber$(Area%) + " Precedes.: "
Variable$ = "data.p[" + LTRIM$(STR$(Area%)) + "].name"
ly% = 8
CASE 5, 9
Area% = (c% - 5) / 4 + 1
Areas% = 2
Quest$ = "          Moment for Name: "
Variable$ = "data.p[" + LTRIM$(STR$(Area%)) + "].moment2"
ly% = 9
CASE 10
Area% = 1
Areas% = 1
Quest$ = " 1st Source: "
Variable$ = "data.source"
ly% = 10
CASE 11, 23
Area% = (c% - 11) / 12 + 1
Areas% = 2
Offset% = -1
Quest$ = "Tolerance " + OrdinaryNumber$(Area%) + " Relation Date: "
Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].tol1"
ly% = 11
CASE 12, 24
Area% = (c% - 12) / 12 + 1
Areas% = 2
Quest$ = "          Which Year? "
Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].date.year"
ly% = 12
CASE 13, 25
Area% = (c% - 13) / 12 + 1
Areas% = 2
Quest$ = "          Which Month? "
Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].date.month"
ly% = 13
CASE 14, 26
Area% = (c% - 14) / 12 + 1
Areas% = 2
Quest$ = "          Which Day? "
Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].date.day"
ly% = 14
CASE 15, 27
Area% = (c% - 15) / 12 + 1
Areas% = 2
Offset% = -2
Quest$ = "          Moment for Relation Name: "
Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].moment1"
ly% = 15
CASE 16, 28
Area% = (c% - 16) / 12 + 1
Areas% = 2
```

```

    Quest$ = OrdinaryNumber$(Area%) + " Rel. Name: "
    Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].name"
    ly% = 16
CASE 17, 29
    Area% = (c% - 17) / 12 + 1
    Areas% = 2
    Quest$ = "                Moment for Name: "
    Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].moment2"
    ly% = 17
CASE 18, 30
    Area% = (c% - 18) / 12 + 1
    Areas% = 2
    Quest$ = "    Tolerance " + OrdinaryNumber$(Area%) + " Duration: "
    Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].tol2"
    ly% = 18
CASE 19, 31
    Area% = (c% - 19) / 12 + 1
    Areas% = 2
    Quest$ = "                How Many Years? "
    Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].duration.year"
    ly% = 19
CASE 20, 32
    Area% = (c% - 20) / 12 + 1
    Areas% = 2
    Quest$ = "                How Many Months? "
    Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].duration.month"
    ly% = 20
CASE 21, 33
    Area% = (c% - 21) / 12 + 1
    Areas% = 2
    Quest$ = "                How Many Days? "
    Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].duration.day"
    ly% = 21
CASE 22, 34
    Area% = (c% - 22) / 12 + 1
    Areas% = 2
    Quest$ = "    " + OrdinaryNumber$(Area% + 1) + " Source: "
    Variable$ = "data.r[" + LTRIM$(STR$(Area%)) + "].source"
    ly% = 22
CASE 35, 36
    Area% = 1
    Areas% = 1
    IF c% = 35 THEN
        Quest$ = "Moments at the Same Time: "
        Offset% = -3
    ELSE
        Quest$ = " "
        Offset% = 15
    END IF
    Variable$ = "data.simultaneous"
    ly% = 23
END SELECT
Buffer$ = Part$(Data$, Variable$)
SELECT CASE c%
CASE 2, 3, 5, 6, 7, 9, 15, 17, 27, 29
    Buffer$ = ToUser$(Buffer$)
CASE 35
    Buffer$ = ToUser$(LEFT$(Buffer$, 3))

```

```

    c% = 35
CASE 36
    Buffer$ = ToUser$(RIGHT$(Buffer$, 2))
    c% = 36
END SELECT
Present Colors%, Quest$, Buffer$, Offset%, "C", Area%, Areas%, ly%

END SUB 'PresentLine _____'

'=====
FUNCTION Question$ (Text$, Default$, InputSuccess%)
'=====
' Will present a question on the screen and wait for Yes or No.
'
' Handling:
' 9/10/2001: Norbert Südland, Munich
' Translation:
' 9/14/2007: Norbert Südland, Aalen
'-----
DIM Answer$
DIM x%
DIM ly%

IF Default$ <> "N" AND Default$ <> "Y" THEN Default$ = "N"
LOCATE 25, 1: PRINT SPACE$(79);
Present 2, Text$ + " [Y/N]", Default$, 0, "C", 1, 1, 25
x% = CSRLIN
ly% = POS(0) - 1
DO
    LOCATE x%, ly%
    Answer$ = Editor$(Default$, 1, 2, InputSuccess%)
    IF InputSuccess% = 1 THEN
        IF Answer$ = "y" THEN Answer$ = "Y"
        IF Answer$ = "n" THEN Answer$ = "N"
    END IF
LOOP UNTIL Answer$ = "Y" OR Answer$ = "N" OR InputSuccess% = 0
PRINT Answer$;
Question$ = Answer$
END FUNCTION 'Question$ _____'

'=====
SUB RestSystem (Sign%, Year%, Month%, Day%)
'=====
' Will clarify a row of date numbers to get an unequivocal situation.
'
' Handling:
' 8/ 4/2001 - 3/27/2003: Norbert Südland
' Translation:
' 11/19/2007: Norbert Südland
'-----
DIM days& 'AS LONG

days& = Sign% * Year% * 360& + Month% * 30& + Day%
Year% = INT(days& / 360&)
days& = days& - Year% * 360&
Month% = INT(days& / 30&)

```

```

Day% = days& - Month% * 30&
IF Year% < 0 THEN
    Year% = -Year%
    Sign% = -1
ELSE
    Sign% = 1
END IF
IF ABS(Year%) >= 10000 THEN ERROR 106
END SUB 'RestSystem _____'

'=====
SUB Reverse (s1%, y1%, m1%, d1%, s2%, y2%, m2%, d2%, Direction$)
'=====
' If 'Direction$' = "-", then the date 'DateValue$' will be mirrored.
' 'Direction' = "+" causes no effect.
'
' Handling:
' 9/ 4/2001: Norbert Südland, Munich
' Translation:
' 11/19/2007: Norbert Südland
'-----

DIM c%          'AS INTEGER
DIM s%
DIM y%
DIM m%
DIM d%

IF Direction$ = "-" THEN
    FOR c% = 1 TO 2
        IF c% = 1 THEN
            s% = s1%: y% = y1%: m% = m1%: d% = d1%
        ELSE
            s% = s2%: y% = y2%: m% = m2%: d% = d2%
        END IF
        s% = -s%
        m% = -m%
        d% = -d%
        RestSystem s%, y%, m%, d%
        IF c% = 1 THEN
            s1% = s%: y1% = y%: m1% = m%: d1% = d%
        ELSE
            s2% = s%: y2% = y%: m2% = m%: d2% = d%
        END IF
    NEXT c%
END IF
END SUB 'Reverse _____'

'=====
FUNCTION SetUp% (Data$, Position&, HistoricChoice%, InputSuccess%)
'=====
'
' Handling:
' 8/ 5/2001 - 9/ 4/2001: Norbert Südland, Munich
' Translation:
' 11/19/2007: Norbert Südland
'-----

DIM Text$

```

```

DIM Answer$
DIM c%

CLS
IF Position& >= 100 THEN
    Text$ = LTRIM$(STR$(INT(Position& / 100)))
    IF INT((Position& MOD 100) / 10) = 0 THEN
        Text$ = Text$ + "0"
    END IF
ELSE
    Text$ = ""
END IF
Text$ = Text$ + OrdinaryNumber$(Position& MOD 100) + " File Record:"
Present 2, Text$, "", 0, "C", 1, 1, 1
IF HistoricChoice% = 2 THEN
    Present 1, "", "(2)...Begin /Enlarge the Data File:", 0, "C", 1, 1, 2
ELSEIF HistoricChoice% = 5 THEN
    Present 1, "", "(5)...Correct a File Record:", 0, "C", 1, 1, 2
END IF
FOR c% = 1 TO 36
    PresentLine Data$, c%, 1
NEXT c%
DO
    FOR c% = 1 TO 35
        c% = SingleInput$(Data$, c%, 2, InputSuccess%)
        IF InputSuccess% = 0 THEN c% = 35
    NEXT c%
    IF InputSuccess% = 1 THEN
        IF c% = 39 THEN c% = SingleInput$(Data$, 34, 2, InputSuccess%)
        IF InputSuccess% = 1 AND c% < 40 THEN
            Answer$ = Question$("Correct the Data?", "N", InputSuccess%)
        END IF
    END IF
    LOOP UNTIL c% = 40 OR Answer$ = "N" OR InputSuccess% = 0
    SetUp% = c% MOD 40
END FUNCTION 'SetUp% _____

'=====
FUNCTION SimultaneousnessCheck$ (Text$)
'=====
' Will correct `Text$` to give sensible entries of simultaneous moments.
' `Text$` must own the length of 5 characters!
'
' Handling:
' 9/ 4/2001 - 12/24/2002:    Norbert Südland
' Translation:
' 11/19/2007 - 11/22/2007:    Norbert Südland
'-----
DIM a$(2)      'AS STRING
DIM empty$
DIM c%         'AS INTEGER
DIM ly%
DIM x%(2)
DIM m%

FOR c% = 1 TO 2

```

```

a$(c%) = SPACE$(6)
x%(c%) = 0
FOR ly% = 1 + (c% - 1) * 3 TO 3 + (c% - 1) * 2
  m% = Moment$(MID$(Text$, ly%, 1))
  IF m% <> 0 THEN
    IF MID$(a$(c%), m%, 1) = " " THEN
      MID$(a$(c%), m%, 1) = MID$(Text$, ly%, 1)
      x%(c%) = x%(c%) + 1
    END IF
  END IF
NEXT ly%

'Has Just a Single Character been Typed in?
'-----

IF x%(c%) < 2 THEN
  a$(c%) = SPACE$(6)
  x%(c%) = 0
END IF
NEXT c%

'Reduction to Fitting Size:
'-----

FOR c% = 1 TO 2
  a$(c%) = RTRIM$(LTRIM$(a$(c%)))
  IF LEN(a$(c%)) > 0 THEN
    IF MID$(a$(c%), 2, 1) = " " THEN 'at least a further character rhs!
      a$(c%) = LEFT$(a$(c%), 1) + LTRIM$(RIGHT$(a$(c%), LEN(a$(c%)) - 1))
    END IF
    IF LEN(a$(c%)) > 2 THEN
      IF MID$(a$(c%), 3, 1) = " " THEN
        a$(c%) = LEFT$(a$(c%), 2) + LTRIM$(RIGHT$(a$(c%), LEN(a$(c%)) - 2))
      END IF
    END IF
  END IF
  a$(c%) = a$(c%) + SPACE$((4 - c%) - LEN(a$(c%)))
NEXT c%

'Comparison of the Combinations:
'-----

'There are 8 combinations of Moments possible in principle:
'"*Ba", "oE+", "*B", "Ba", "aE", "oE", "o+", "E+".
'The following pairs of combinations are valid:
'
'


|    | *B | Ba | aE | oE | o+ | E+ |
|----|----|----|----|----|----|----|
| *B | O  | O  | X  | X  | X  | -  |
| Ba | O  | O  | -  | -  | X  | -  |
| aE | X  | -  | O  | -  | X  | -  |
| oE | X  | -  | -  | O  | O  | O  |
| o+ | X  | X  | X  | O  | O  | O  |
| E+ | -  | -  | -  | O  | O  | O  |


'
'The concrete symbols mean:
'
' O possibility of reduction to something sensible
' X valid combination
' - invalid combination or reduction
'-----
empty$ = SPACE$(2)

```

```

SELECT CASE a$(1)                                'Changes of order are rather perturbing!
CASE "*"B" + CHR$(224), CHR$(234) + "E+"
  a$(2) = empty$
CASE "*"B" + SPACE$(1)
  SELECT CASE a$(2)
  CASE "B" + CHR$(224)
    a$(1) = "*"B" + CHR$(224): a$(2) = empty$
  CASE CHR$(224) + "E", CHR$(234) + "E", CHR$(234) + "+"
  CASE ELSE
    a$(2) = empty$
  END SELECT
CASE "B" + CHR$(224) + SPACE$(1)
  SELECT CASE a$(2)
  CASE "*"B"
    a$(1) = "*"B" + CHR$(224): a$(2) = empty$
  CASE CHR$(234) + "+"
  CASE ELSE
    a$(2) = empty$
  END SELECT
CASE CHR$(224) + "E" + SPACE$(1)
  SELECT CASE a$(2)
  CASE "*"B"
    a$(1) = "*"B ": a$(2) = CHR$(224) + "E"
  CASE CHR$(234) + "+"
  CASE ELSE
    a$(2) = empty$
  END SELECT
CASE CHR$(234) + "E" + SPACE$(1)
  SELECT CASE a$(2)
  CASE "*"B"
    a$(1) = "*"B ": a$(2) = CHR$(234) + "E"
  CASE CHR$(234) + "+", "E+"
    a$(1) = CHR$(234) + "E+": a$(2) = empty$
  CASE ELSE
    a$(2) = empty$
  END SELECT
CASE CHR$(234) + "+" + SPACE$(1)
  SELECT CASE a$(2)
  CASE "*"B"
    a$(1) = "*"B ": a$(2) = CHR$(234) + "+"
  CASE "B" + CHR$(224)
    a$(1) = "B" + CHR$(224) + SPACE$(1): a$(2) = CHR$(234) + "+"
  CASE CHR$(224) + "E"
    a$(1) = CHR$(224) + "E ": a$(2) = CHR$(234) + "+"
  CASE CHR$(234) + "E", "E+"
    a$(1) = CHR$(234) + "E+": a$(2) = empty$
  CASE ELSE
    a$(2) = empty$
  END SELECT
CASE "E+ "
  SELECT CASE a$(2)
  CASE CHR$(234) + "E", CHR$(234) + "+"
    a$(1) = CHR$(234) + "E+": a$(2) = empty$
  CASE ELSE
    a$(2) = empty$
  END SELECT
CASE ELSE
  SELECT CASE a$(2)

```

```

CASE "*"B", "B" + CHR$(224), CHR$(224) + "E"
a$(1) = a$(2) + SPACE$(1): a$(2) = empty$
CASE CHR$(234) + "E", CHR$(234) + "+", "E+"
a$(1) = a$(2) + SPACE$(1): a$(2) = empty$
CASE ELSE
a$(2) = empty$: a$(1) = a$(2) + SPACE$(1)
END SELECT
END SELECT

'Result:
'-----'

SimultaneousnessCheck$ = a$(1) + a$(2)
END FUNCTION 'SimultaneousnessCheck$ _____'

'=====
FUNCTION SingleInput% (Data$, c%, Colors%, InputSuccess%)
'=====
' Deals with the concrete input of a file record.
'
' Handling:
' 8/ 5/2001 - 9/ 4/2001:    Norbert Südland, Munich
' 9/ 6/2002 - 9/26/2003:    Norbert Südland, Aalen
' 4/18/2007 - 11/22/2007:    Norbert Südland, Aalen
' Translation:
' 11/20/2007 - 11/22/2007:    Norbert Südland
'-----'

DIM K1$      'AS STRING
DIM K2$
DIM Variable$
DIM Buffer$
DIM Simultaneous$
DIM source$
DIM a$
DIM l%      'AS INTEGER
DIM x%
DIM ly%
DIM counter%
DIM addition%
DIM Length%
DIM completed%
DIM DeleteRelation2%
DIM AddSource%
DIM y%
DIM m%
DIM d%
DIM start%
DIM finish%
DIM f%
DIM SimultaneousSource%

l% = LEN("data.r[?]")      'needed for truncating Variable$
SELECT CASE c%
CASE 1
K1$ = CHR$(34) + "Name /Event" + CHR$(34) + " asks for the name "
K1$ = K1$ + "of a person or event, "
K2$ = "which is dated within a historical background and its "
K2$ = K2$ + "corresponding field."

```

```

CASE 2, 5, 6, 9, 15, 17, 27, 29, 35
  K1$ = "*" , B , " + CHR$(224) + " , " + CHR$(234) + " , E , + as a "
  K1$ = K1$ + "symbol for the corresponding moment means:"
  K2$ = "begin of life, of 1st, of 2nd periode, end of 2nd, of 1st "
  K2$ = K2$ + "periode, of life"
CASE 3, 7
  K1$ = CHR$(34) + "Directly Before?" + CHR$(34) + " asks, how exact "
  K1$ = K1$ + "the relative dating is, i.e., whether"
  K2$ = CHR$(34) + "Name /Event" + CHR$(34) + " follows "
  K2$ = K2$ + "without a time gap after the " + CHR$(34)
  K2$ = K2$ + "predecessor" + CHR$(34) + " , or not."
CASE 4, 8
  K1$ = CHR$(34) + "Predecessor" + CHR$(34) + " asks for the predecessor "
  K1$ = K1$ + "or pre-event of " + CHR$(34) + "Name /Event" + CHR$(34) + "."
CASE 10, 22, 34
  K1$ = CHR$(34) + "Source" + CHR$(34) + " is used as a documentary help "
  K1$ = K1$ + "to be able to check"
  K2$ = "the presented relation structure by the historical sources "
  K2$ = K2$ + "in detail."
CASE 11, 18, 23, 30
  IF c% = 11 OR c% = 23 THEN
    K1$ = "The relation date counts time starting by the moment for "
    K1$ = K1$ + CHR$(34) + "Relation Name" + CHR$(34) + "."
  ELSE
    K1$ = "The duration corresponds to the presented moment of "
    K1$ = K1$ + CHR$(34) + "Name" + CHR$(34) + "."
  END IF
  K2$ = "< , = , ± , > as tolerance symbol means: ord. number, without, "
  K2$ = K2$ + "normal, limited tol."
CASE 12 TO 14, 19 TO 21, 24 TO 26, 31 TO 33
  K1$ = "Calculation in Years of 12 Months à 30 Days"
  K2$ = "Valid input area: 1"
  SELECT CASE c%
    CASE 12, 24
      K2$ = K2$ + ".year until 9999.year - 0.year means deleting."
    CASE 13, 25
      K2$ = K2$ + ".month until 12.month - 0.month means deleting."
    CASE 14, 26
      K2$ = K2$ + ".day until 30.day - 0.day means deleting."
    CASE 19, 31
      K2$ = K2$ + " year until 9999 years - 0 years means deleting."
    CASE 20, 32
      K2$ = K2$ + " month until 12 months - 0 months means deleting."
    CASE 21, 33
      K2$ = K2$ + " day until 30 days - 0 days means deleting."
  END SELECT
CASE 16, 28
  K1$ = CHR$(34) + "Relation Name" + CHR$(34) + " asks for a person or"
  K1$ = K1$ + " an event corresponding to the moment,"
  K2$ = "since when the presented relation date is counted."
END SELECT
Present 2, K1$, "", 0, "C", 1, 1, 3
Present 2, K2$, "", 0, "C", 1, 1, 4
PresentLine Data$, c%, 2
x% = CSRLIN
ly% = POS(0) - 1
addition% = 0
SELECT CASE c%

```

```

CASE 1, 4, 8, 10, 16, 22, 28, 34
  Length% = SIZEOF%("NAME")
  ly% = ly% - Length% + 1
  SELECT CASE c%
  CASE 1
    Variable$ = "data.name"
  CASE 4, 8
    Variable$ = "data.p[" + LTRIM$(STR$(c% / 4)) + "].name"
  CASE 10
    Variable$ = "data.source"
  CASE 16, 28
    Variable$ = "data.r[" + LTRIM$(STR$((c% - 16) / 12 + 1)) + "].name"
  CASE 22, 34
    Variable$ = "data.r[" + LTRIM$(STR$((c% - 22) / 12 + 1)) + "].source"
  END SELECT
  Buffer$ = Part$(Data$, Variable$)
  DO
    LOCATE 25, 1: PRINT SPACE$(79);
    LOCATE x%, ly%
    Buffer$ = Editor$(Buffer$, Length%, 3, InputSuccess%)
    IF InputSuccess% = 1 AND c% = 1 AND Buffer$ = SPACE$(Length%) THEN
      IF Question$("Delete Record?", "N", InputSuccess%) = "Y" THEN
        c% = 39
      END IF
    END IF
    LOOP UNTIL Buffer$ <> SPACE$(Length%) OR c% = 39 OR InputSuccess% = 0
    IF InputSuccess% = 0 THEN GOTO EndSingleInput
    IF c% < 39 THEN Change Data$, Variable$, Buffer$
  CASE 2, 5, 6, 9, 15, 17, 27, 29
    SELECT CASE c%
    CASE 2, 6
      Variable$ = "data.p[" + LTRIM$(STR$((c% - 2) / 4 + 1)) + "].moment1"
    CASE 5, 9
      Variable$ = "data.p[" + LTRIM$(STR$((c% - 5) / 4 + 1)) + "].moment2"
    CASE 15, 27
      Variable$ = "data.r[" + LTRIM$(STR$((c% - 15) / 12 + 1)) + "].moment1"
    CASE 17, 29
      Variable$ = "data.r[" + LTRIM$(STR$((c% - 17) / 12 + 1)) + "].moment2"
    END SELECT
    Buffer$ = ToUser$(Part$(Data$, Variable$))
  DO
    LOCATE x%, ly%
    Buffer$ = Editor$(Buffer$, 1, 3, InputSuccess%)
    IF InputSuccess% = 0 THEN GOTO SingleInputEndLoop1
    SELECT CASE Buffer$
    CASE "*", "B", CHR$(224), CHR$(234), "E", "+"
      completed% = 1
    CASE CHR$(197)
      'large +
      Buffer$ = "+"
      completed% = 1
    CASE "a", "A"
      Buffer$ = CHR$(224)
      'alpha
      completed% = 1
    CASE "o", "O"
      Buffer$ = CHR$(234)
      'Omega
      completed% = 1
    CASE "b", "e"
      Buffer$ = UCASE$(Buffer$)

```

```

        completed% = 1
    END SELECT
    IF completed% = 1 THEN Change Data$, Variable$, ToFile$(Buffer$)
    IF Buffer$ = " " THEN
        DeleteRelation2% = 0
        SELECT CASE c%
        CASE 2
            completed% = 1
            Change Data$, "data.p[1]", SPACE$(Length% = SIZEOF("predecessor"))
            Change Data$, "data.p[2]", SPACE$(Length%)

            'Check, Whether a Pre-Dating is Available:
            '-----'
            AddSource% = 0
            FOR counter% = 1 TO 6
                Variable$ = "data.date[" + LTRIM$(STR$(counter%)) + "].status"
                Buffer$ = Part$(Data$, Variable$)
                SELECT CASE Buffer$
                CASE CHR$(219), CHR$(221), CHR$(222)          'several blocks
                    AddSource% = 1
                    counter% = 6
                END SELECT
            NEXT counter%

            'Check, Whether the Characters " = " Occur Within the Name:
            '-----'
            IF INSTR(Part$(Data$, "data.name"), " = ") > 0 THEN
                AddSource% = 1
            END IF

            IF AddSource% = 1 THEN
                addition% = 7
            ELSE
                Change Data$, "data.source", SPACE$(Length%)
                addition% = 8
            END IF
        CASE 6
            completed% = 1
            Change Data$, "data.p[2]", SPACE$(SIZEOF("predecessor"))
            addition% = 3
        CASE 17
            IF Part$(Data$, LEFT$(Variable$, 1%) + ".tol1") = " " THEN
                completed% = 1
                Change Data$, "data.r[1]", SPACE$(SIZEOF("relation"))
                DeleteRelation2% = 1
                addition% = 17
            END IF
        CASE 29
            IF Part$(Data$, LEFT$(Variable$, 1%) + ".tol1") = " " THEN
                completed% = 1
                DeleteRelation2% = 1
                addition% = 5
            END IF
        END SELECT
    IF DeleteRelation2% = 1 THEN

        'Check, Whether Simultaneous Events are Existent:
        '-----'

```

```

        Length% = SIZEOF$("relation")
        Simultaneous$ = Part$(Data$, "data.simultaneous")
        IF Simultaneous$ = SPACE$(LEN(Simultaneous$)) THEN
            Change Data$, "data.r[2]", SPACE$(Length%)
        ELSE
            source$ = Part$(Data$, "data.r[2].source")
            Change Data$, "data.r[2]", SPACE$(Length%)
            Change Data$, "data.r[2].source", source$
        END IF
    END IF
END IF
END IF
'=====
SingleInputEndLoop1:
'=====
    LOOP UNTIL completed% = 1 OR InputSuccess% = 0
CASE 3, 7
    Variable$ = "data.p[" + LTRIM$(STR$((c% - 3) / 4 + 1)) + "].direct"
    Buffer$ = ToUser$(Part$(Data$, Variable$))
    DO
        LOCATE x%, ly%
        Buffer$ = Editor$(Buffer$, 1, 3, InputSuccess%)
        IF InputSuccess% = 0 THEN GOTO SingleInputEndLoop2
        SELECT CASE Buffer$
        CASE "y", "Y", "n", "N"
            completed% = 1
            Change Data$, Variable$, ToFile$(UCASE$(Buffer$))
        END SELECT
    END IF
'=====
SingleInputEndLoop2:
'=====
    LOOP UNTIL completed% = 1 OR InputSuccess% = 0
CASE 11, 18, 23, 30
    SELECT CASE c%
    CASE 11, 23
        Variable$ = "data.r[" + LTRIM$(STR$((c% - 11) / 12 + 1)) + "].tol1"
    CASE 18, 30
        Variable$ = "data.r[" + LTRIM$(STR$((c% - 18) / 12 + 1)) + "].tol2"
    END SELECT
    Buffer$ = Part$(Data$, Variable$)
    DO
        LOCATE x%, ly%
        Buffer$ = Editor$(Buffer$, 1, 3, InputSuccess%)
        IF InputSuccess% = 0 THEN GOTO SingleInputEndLoop3
        SELECT CASE Buffer$
        CASE "<", "=", "±", ">", " ", "+", "-", CHR$(242), CHR$(243), CHR$(240)
            SELECT CASE Buffer$
            CASE CHR$(243) ' <=
                Buffer$ = "<"
            CASE CHR$(240) ' ==
                Buffer$ = "="
            CASE "+", "- "
                Buffer$ = "±"
            CASE CHR$(242) ' >=
                Buffer$ = ">"
            END SELECT
        END SELECT
        completed% = 1
    END SELECT
    IF completed% = 1 THEN Change Data$, Variable$, Buffer$

```

```

IF Buffer$ = " " THEN
  IF c% = 11 OR c% = 23 THEN
    Variable$ = LEFT$(Variable$, 1%)
    Length% = SIZEOF$("time")
    Change Data$, Variable$ + ".date", SPACE$(Length%)
    Change Data$, Variable$ + ".moment1", Buffer$
    Change Data$, Variable$ + ".name", SPACE$(SIZEOF$("NAME"))
    addition% = 5
  ELSEIF c% = 18 OR c% = 30 THEN
    Variable$ = LEFT$(Variable$, 1%)
    IF Part$(Data$, Variable$ + ".tol1") = " " THEN
      Change Data$, Variable$ + ".moment2", Buffer$
      Length% = SIZEOF$("NAME")
      Change Data$, Variable$ + ".source", SPACE$(Length%)
      IF c% = 18 THEN
        Change Data$, "data.r[2]", SPACE$(SIZEOF$("relation"))
        addition% = 12
      END IF
      c% = c% - 1
      addition% = addition% + 2
    END IF
    Change Data$, Variable$ + ".tol2", Buffer$
    Length% = SIZEOF$("time")
    Change Data$, Variable$ + ".duration", SPACE$(Length%)
    addition% = addition% + 3
  END IF
END IF
'=====
SingleInputEndLoop3:
'=====
  LOOP UNTIL completed% = 1 OR InputSuccess% = 0
  CASE 12 TO 14, 19 TO 21, 24 TO 26, 31 TO 33
    Variable$ = "data.r[" + LTRIM$(STR$(INT(c% / 23) + 1)) + "]"
    SELECT CASE c%
    CASE 12, 24
      ly% = ly% - 3
      Variable$ = Variable$ + ".date"
      DO
        Buffer$ = Part$(Data$, Variable$ + ".year")
        LOCATE x%, ly%
        Buffer$ = Editor$(Buffer$, 4, 1, InputSuccess%)
        IF InputSuccess% = 1 THEN y% = VAL(Buffer$)
      LOOP UNTIL y% >= 0 OR InputSuccess% = 0
      IF y% = 0 THEN
        Buffer$ = SPACE$(4)
        Change Data$, Variable$ + ".ys", SPACE$(3)
      ELSE
        Change Data$, Variable$ + ".ys", ".Y "
      END IF
      Variable$ = Variable$ + ".year"
    CASE 13, 25
      ly% = ly% - 1
      Variable$ = Variable$ + ".date"
      DO
        Buffer$ = Part$(Data$, Variable$ + ".month")
        LOCATE x%, ly%
        Buffer$ = Editor$(Buffer$, 2, 1, InputSuccess%)
        IF InputSuccess% = 1 THEN m% = VAL(Buffer$)

```

```
LOOP UNTIL m% >= 0 AND m% <= 12 OR InputSuccess% = 0
IF m% = 0 THEN
    Buffer$ = SPACE$(2)
    Change Data$, Variable$ + ".ms", SPACE$(3)
ELSE
    Change Data$, Variable$ + ".ms", ".M "
END IF
Variable$ = Variable$ + ".month"
CASE 14, 26
    ly% = ly% - 1
    Variable$ = Variable$ + ".date"
DO
    Buffer$ = Part$(Data$, Variable$ + ".day")
    LOCATE x%, ly%
    Buffer$ = Editor$(Buffer$, 2, 1, InputSuccess%)
    IF InputSuccess% = 1 THEN d% = VAL(Buffer$)
    LOOP UNTIL d% >= 0 AND d% <= 30 OR InputSuccess% = 0
    IF d% = 0 THEN
        Buffer$ = SPACE$(2)
        Change Data$, Variable$ + ".ds", SPACE$(2)
    ELSE
        Change Data$, Variable$ + ".ds", ".D"
    END IF
    Variable$ = Variable$ + ".day"
CASE 19, 31
    ly% = ly% - 3
    Variable$ = Variable$ + ".duration"
DO
    Buffer$ = Part$(Data$, Variable$ + ".year")
    LOCATE x%, ly%
    Buffer$ = Editor$(Buffer$, 4, 1, InputSuccess%)
    IF InputSuccess% = 1 THEN y% = VAL(Buffer$)
    LOOP UNTIL y% >= 0 OR InputSuccess% = 0
    IF y% = 0 THEN
        Buffer$ = SPACE$(4)
        Change Data$, Variable$ + ".ys", SPACE$(3)
    ELSE
        Change Data$, Variable$ + ".ys", " Y "
    END IF
    Variable$ = Variable$ + ".year"
CASE 20, 32
    ly% = ly% - 1
    Variable$ = Variable$ + ".duration"
DO
    Buffer$ = Part$(Data$, Variable$ + ".month")
    LOCATE x%, ly%
    Buffer$ = Editor$(Buffer$, 2, 1, InputSuccess%)
    IF InputSuccess% = 1 THEN m% = VAL(Buffer$)
    LOOP UNTIL m% >= 0 AND m% <= 12 OR InputSuccess% = 0
    IF m% = 0 THEN
        Buffer$ = SPACE$(2)
        Change Data$, Variable$ + ".ms", SPACE$(3)
    ELSE
        Change Data$, Variable$ + ".ms", " M "
    END IF
    Variable$ = Variable$ + ".month"
CASE 21, 33
    ly% = ly% - 1
```

```

Variable$ = Variable$ + ".duration"
DO
  Buffer$ = Part$(Data$, Variable$ + ".day")
  LOCATE x%, ly%
  Buffer$ = Editor$(Buffer$, 2, 1, InputSuccess%)
  IF InputSuccess% = 1 THEN d% = VAL(Buffer$)
  LOOP UNTIL d% >= 0 AND d% <= 30 OR InputSuccess% = 0
  IF d% = 0 THEN
    Buffer$ = SPACE$(2)
    Change Data$, Variable$ + ".ds", SPACE$(2)
  ELSE
    Change Data$, Variable$ + ".ds", " D"
  END IF
  Variable$ = Variable$ + ".day"
END SELECT
IF InputSuccess% = 1 THEN Change Data$, Variable$, Buffer$
CASE 35
  Simultaneous$ = ToUser$(Part$(Data$, "data.simultaneous"))
  ly% = ly% - 2
  start% = 1
  finish% = 3
  DO
    LOCATE x%, ly%
    Length% = finish% - start% + 1
    Buffer$ = MID$(Simultaneous$, start%, Length%)
    Buffer$ = Editor$(Buffer$, Length%, 3, InputSuccess%)
    IF InputSuccess% = 1 THEN
      FOR f% = 1 TO Length%
        a$ = MID$(Buffer$, f%, 1)
        SELECT CASE a$
          CASE "B", "b", CHR$(224), CHR$(234), "E", "+" '., alpha, Omega, ..
          CASE CHR$(197) 'large +
            MID$(Buffer$, f%, 1) = "+"
          CASE "a", "A"
            MID$(Buffer$, f%, 1) = CHR$(224) 'alpha
          CASE "o", "O"
            MID$(Buffer$, f%, 1) = CHR$(234) 'Omega
          CASE "b", "e"
            MID$(Buffer$, f%, 1) = UCASE$(a$)
        END SELECT
      NEXT f%
      MID$(Simultaneous$, start%, Length%) = Buffer$
      Simultaneous$ = SimultaneousnessCheck$(Simultaneous$)
      IF LEFT$(Simultaneous$, 1) = " " THEN
        completed% = 1
      ELSE
        IF start% = 1 THEN
          start% = 4: finish% = 5: ly% = ly% + 6
          Change Data$, "data.simultaneous", ToFile$(Simultaneous$)
          PresentLine Data$, 35, 1
        ELSE
          completed% = 1
        END IF
      END IF
    END IF
  LOOP UNTIL completed% = 1 OR InputSuccess% = 0
  IF InputSuccess% = 1 THEN
    Change Data$, "data.simultaneous", ToFile$(Simultaneous$)
  
```

```

IF Simultaneous$ = SPACE$(5) THEN

    'Check, Whether the 3rd Source must be Deleted:
    '-----'
    IF Part$(Data$, "data.r[2].tol1") = SPACE$(1) THEN
        IF Part$(Data$, "data.r[2].moment2") = SPACE$(1) THEN
            source$ = SPACE$(SIZEOF("data.r[2].source"))
            Change Data$, "data.r[2].source", source$
            PresentLine Data$, 34, 1
        END IF
    END IF
ELSE
    SimultaneousSource% = 1
END IF
addition% = 1
END IF
END SELECT
IF InputSuccess% = 0 THEN GOTO EndSingleInput
LOCATE 3, 1
PRINT SPACE$(160);
IF c% < 39 THEN
    FOR x% = c% TO c% + addition%
        PresentLine Data$, x%, 1
    NEXT x%
    c% = x% - 1
END IF
IF SimultaneousSource% = 1 THEN c% = 38

'=====
EndSingleInput:
'=====
    SingleInput% = c%
END FUNCTION 'SingleInput% _____

'=====
FUNCTION SIZEOF% (Variable$)
'=====
    ' Will give the length of the data structure being named 'Variable$'.
    '
    ' Handling:
    ' 9/ 4/2001: Norbert Südland, Munich
    ' 9/ 6/2002: Norbert Südland, Aalen
    '
    ' Translation:
    ' 11/24/2007: Norbert Südland
    '-----'

DIM seek$           'AS STRING
DIM FurtherSeek$
DIM Length%         'AS INTEGER
DIM p%
DIM finish%
DIM where%
DIM result%

Length% = LEN(Variable$)
p% = STRLEN%(Variable$, ".")
IF p% = Length% THEN

```

```

    seek$ = Variable$
ELSE
    p% = p% + 1
    p% = p% + STRLEN$(MID$(Variable$, p% + 1, Length% - p%), ".")
    IF p% < Length% THEN
        seek$ = LEFT$(Variable$, p%)
    ELSE
        seek$ = Variable$
    END IF
END IF
finish% = GlobVarNumber%

where% = Number%(seek$, 1, finish%)
IF where% = 0 THEN
    PRINT
    PRINT "The data structure " + CHR$(34) + Variable$; CHR$(34);
    PRINT " is missing."
801 ERROR 9
ELSE
    result% = GVLength%(where%)
    IF p% < Length% THEN
        FurtherSeek$ = GVType$(where%) + MID$(Variable$, p% + 1, Length% - p%)
        result% = SIZEOF$(FurtherSeek$)
    END IF
END IF

SIZEOF% = result%
END FUNCTION 'SIZEOF% _____'

'=====
SUB StartingPage (Answer%, InputSuccess%)
'=====
' Will present the starting page and control the corresponding bar menu.
'
' Handling:
' 9/ 4/2001 - 9/14/2001: Norbert Südland, Munich
' 8/31/2002 - 12/ 4/2002: Norbert Südland, Aalen
' 1/20/2009:          Norbert Südland, Aalen
' 8/18/2012:          Norbert Südland, Aalen
' 10/21/2022:         Norbert Südland, Aalen
'
' Translation:
' 11/24/2007:          Norbert Südland
'-----

DIM Text$      'AS STRING
DIM result$
DIM c%         'AS INTEGER
DIM before%

CLS
Present 1, "H I S T O R I C", "", 0, "C", 1, 1, 1
Text$ = "Program for Generating Historical Time Tables"
Present 1, Text$, "", 0, "C", 1, 1, 2
Text$ = "Calculation in Years of 12 Months à 30 Days"
Present 1, Text$, "", 0, "C", 1, 1, 3
Present 1, VersionProgrammingDate$, "", 0, "C", 1, 1, 4
Text$ = "(C) 1985-" + RIGHT$(VersionProgrammingDate$, 4)

```

```

Text$ = Text$ + " Norbert Südland, Eckhard Walter"
Present 1, Text$, "", 0, "C", 1, 1, 5
Text$ = "Today's Date (Gregorianic Calendar Reform of the Christian"
Text$ = Text$ + " Chronology):"
Present 1, Text$, "", 0, "C", 1, 1, 7
Text$ = CurrentDate$
Present 1, Text$, "", 0, "C", 1, 1, 8

'Bar Menu:
'-----'
FOR c% = 1 TO 10
    Text$ = StartingText$(c% MOD 10)
    IF (c% - Answer%) MOD 10 = 0 THEN
        Present 1, "", Text$, 0, "C", 1, 1, 9 + c%
    ELSE
        Present 1, Text$, "", 0, "C", 1, 1, 9 + c%
    END IF
NEXT c%

'Bottom Line:
'-----'
IF ENVIRON$("COMSPEC") = "Z:\COMMAND.COM" THEN
    Text$ = " [ Ctrl ]-[ Scroll ] "      'DOSBox 0.74
ELSEIF ENVIRON$("DOSDIR") <> "" THEN
    Text$ = " [ Ctrl ]-[ Scroll ] "      'FreeDOS 1.3
ELSE
    Text$ = " [ Ctrl ]-[ Pause ] "
END IF
Present 2, "View Program Code:", Text$, 0, "C", 1, 1, 25

DO
    Text$ = "Your Choice: "
    Present 1, Text$, "", 0, "C", 1, 1, 21
    InputSuccess% = 1
    before% = Answer%
    result$ = Editor$(LTRIM$(STR$(Answer%)), 1, 1, InputSuccess%)
    Answer% = VAL(result$)
    IF InputSuccess% = 0 AND Answer% = before% THEN
        IF Answer% > 0 THEN Answer% = (Answer% + 1) MOD 10
        result$ = LTRIM$(STR$(Answer%))
    END IF
    PRINT RIGHT$(result$, 1)
    Text$ = " " + StartingText$(before%) + " "
    IF before% = 0 THEN before% = 10
    Present 1, Text$, "", 0, "C", 1, 1, 9 + before%
    Text$ = StartingText$(Answer%)
    IF Answer% = 0 THEN Answer% = 10
    Present 1, "", Text$, 0, "C", 1, 1, 9 + Answer%
    Answer% = Answer% MOD 10
LOOP UNTIL InputSuccess% = 1 OR (before% = 10 AND Answer% = 0)
END SUB 'StartingPage _____'

'=====
FUNCTION StartingText$ (Answer%)
'=====
' Will give the text of the starting bar menu.
'
' Handling:

```

```

' 09/04/2001 - 09/08/2001: Norbert Südland, Munich
' Translation:
' 09/14/2007: Norbert Südland, Aalen
'-----
DIM Text$

SELECT CASE Answer% MOD 10
CASE 1
Text$ = "(1)...Change the Current Date      "
CASE 2
Text$ = "(2)...Begin /Enlarge the Data File"
CASE 3
Text$ = "(3)...Delete a Data Record        "
CASE 4
Text$ = "(4)...Pre-Date a Data Record      "
CASE 5
Text$ = "(5)...Overview and Correction     "
CASE 6
Text$ = "(6)...Calculation                  "
CASE 7
Text$ = "(7)...Sort the Data File          "
CASE 8
Text$ = "(8)...Show /Transform the Result  "
CASE 9
Text$ = "(9)...Print the Data File         "
CASE 0
Text$ = "(0)...Terminate the Program       "
END SELECT
StartingText$ = Text$
END FUNCTION 'StartingText$ _____

'=====
FUNCTION STRLEN% (Text$, EndCharacter$)
'=====
' Will find the string length of `Text$` until to the first occurrence of
' all `EndCharacter$`, like being usual within the C programming language.
' If `EndCharacter$` does not occur, the string length `LEN(Text$)` is
' returned.
'
' Handling:
' 8/ 4/2001: Norbert Südland, Munich
' Check:
' 8/ 4/2001: Norbert Südland, Munich
' Translation:
' 11/24/2007: Norbert Südland, Aalen
'-----
DIM IntermediateResult% 'AS INTEGER

IntermediateResult% = INSTR(Text$, EndCharacter$)
IF IntermediateResult% = 0 THEN
IntermediateResult% = LEN(Text$)
ELSE
IntermediateResult% = IntermediateResult% - 1
END IF

STRLEN% = IntermediateResult%
END FUNCTION 'STRLEN% _____

```

```

'=====
SUB TermToNumber (Term$, Sign%, Year%, Month%, Day%)
'=====
' Will convert a `Term$` to a row of integer numbers.
'
' Translation:
' 11/24/2007: Norbert Suedland
'-----

IF Part$(Term$, "term.sign") = "-" THEN
    Sign% = -1
ELSE
    Sign% = 1
END IF
Year% = VAL(Part$(Term$, "term.year"))
Month% = VAL(Part$(Term$, "term.month"))
Day% = VAL(Part$(Term$, "term.day"))
END SUB 'TermToNumber

'=====
SUB TimeShift (s1%, y1%, m1%, d1%, Status$, s2%, y2%, m2%, d2%, Direction$)
'=====
' Will shift dates with negative year number to plain numbers,
' thus there will be a number zero in date calculation.
' Usually date numbers are ordinary numbers.
'
' Handling:
' 9/ 4/2001: Norbert Suedland, Munich
' Translation:
' 11/24/2007: Norbert Suedland, Aalen
'-----

DIM start% 'AS INTEGER
DIM finish%
DIM c%
DIM s%
DIM y%
DIM m%
DIM d%

SELECT CASE Status$
CASE " "
    start% = 1
    finish% = 2
CASE CHR$(242) ' >=
    start% = 1
    finish% = 1
CASE CHR$(243) ' <=
    start% = 2
    finish% = 2
END SELECT
FOR c% = start% TO finish%
IF c% = 1 THEN
    s% = s1%: y% = y1%: m% = m1%: d% = d1%
ELSE
    s% = s2%: y% = y2%: m% = m2%: d% = d2%
END IF

```

```

IF Direction$ = "+" THEN
    RestSystem s%, y%, m%, d%
    IF s% = 1 THEN
        y% = y% + 1
    END IF
    m% = m% + 1
    d% = d% + 1
ELSE
    IF s% = 1 THEN
        IF y% < 1 THEN ERROR 110
        y% = y% - 1
    END IF
    m% = m% - 1
    d% = d% - 1
    RestSystem s%, y%, m%, d%
END IF
IF c% = 1 THEN
    s1% = s%: y1% = y%: m1% = m%: d1% = d%
ELSE
    s2% = s%: y2% = y%: m2% = m%: d2% = d%
END IF
NEXT c%
END SUB 'TimeShift _____'

'=====
FUNCTION ToFile$ (symbol$)
'=====
' Will change 'Symbol$' to an international code within the data file.
' This function causes also old data files to run correctly.
'
' Handling:
' 11/20/2007 - 11/22/2007: Norbert Südland
' 1/20/2009: Norbert Südland, D-73431 Aalen
'-----
DIM result$
DIM c%

result$ = SPACE$(LEN(symbol$))
FOR c% = 1 TO LEN(symbol$)
    SELECT CASE MID$(symbol$, c%, 1)
        CASE "1", "J", "Y" 'Yes
            MID$(result$, c%, 1) = "1"
        CASE "0", "N" 'No
            MID$(result$, c%, 1) = "0"
        CASE "{", "*" 'Birth
            MID$(result$, c%, 1) = "{"
        CASE "[", "A", "B" 'Begin 1st Periode
            MID$(result$, c%, 1) = "["
        CASE "(", "W", CHR$(224) 'Begin 2nd Periode
            MID$(result$, c%, 1) = "("
        CASE ")", "X", CHR$(234) 'End 2nd Periode
            MID$(result$, c%, 1) = ")"
        CASE "]", "E" 'End 1st Periode
            MID$(result$, c%, 1) = "]"
        CASE "}", "+" 'Death
            MID$(result$, c%, 1) = "}"
        CASE ELSE
            MID$(result$, c%, 1) = SPACE$(1)
    END SELECT
NEXT c%

```

```

    END SELECT
NEXT c%

ToFile$ = result$
END FUNCTION 'ToFile$ _____'

'=====
FUNCTION ToUser$ (symbol$)
'=====
' Will transform the international 'Symbol$' code to an English user code.
' This function will work correctly with an old data file, too.
'
' Handling:
' 11/20/2007 - 11/22/2007: Norbert Südland
'-----
DIM result$
DIM c%

result$ = SPACE$(LEN(symbol$))
FOR c% = 1 TO LEN(symbol$)
    SELECT CASE MID$(symbol$, c%, 1)
        CASE "1", "J", "Y"
            MID$(result$, c%, 1) = "Y"          'Yes
        CASE "0", "N"
            MID$(result$, c%, 1) = "N"          'No
        CASE "{", "*"
            MID$(result$, c%, 1) = "*"          'Birth
        CASE "[", "A", "B"
            MID$(result$, c%, 1) = "B"          'Begin 1st Periode
        CASE "(", "W", CHR$(224)
            MID$(result$, c%, 1) = CHR$(224)    'alpha
            MID$(result$, c%, 1) = CHR$(224)    'Begin 2nd Periode
        CASE ")", "X", CHR$(234)
            MID$(result$, c%, 1) = CHR$(234)    'Omega
            MID$(result$, c%, 1) = CHR$(234)    'End 2nd Periode
        CASE "]", "E"
            MID$(result$, c%, 1) = "E"          'End 1st Periode
        CASE "}", "+"
            MID$(result$, c%, 1) = "+"          'Death
        CASE ELSE
            MID$(result$, c%, 1) = " "
    END SELECT
NEXT c%

ToUser$ = result$
END FUNCTION 'ToUser$ _____'

'=====
FUNCTION WeekDay% (Ctry%, y%, m%, d%)
'=====
' Will calculate the week day for a given date.
'
' Valid values for century (ctry%):          -10 until 100 without 0
' Valid values for year number (y%) :         1 until 100
' Valid values for month number (m%) :        1 until 12
' Valid values for day number (d%) :          1 until 31
' Valid values for the date : 1/ 1/-999 until 12/31/9999
'
' The Given Numbers are Generated to a Sensible Date.
'-----

```

```

' The reference date, which can be choosen arbitrarily, is set to Friday,
' May 14th, 1948, which is the day of Israel's refoundation.
'
' The Gregorianic Calendar Reform owning a time jump from Thursday,
' October 4th in 1582 to Friday, October 15th in 1582, is taken into
' consideration.
' ([Zem1987], Introduction and Section 2.65, pages 11 and 28-29)
'
' The Julian Calendar Reform from 45 before Christ by Caius Julius Caesar
' ([Zem1987], section 2.64, page 27) is taken into consideration.
'
' Before the Julian Calendar Reform the Egyptian sun year owning exactly
' 365 days is used, which counts 1460 years to be already 1461 years.
'
' Thus the comparison with historical dates can differ and may need a
' transformation.
'
' For checking purpose now some undisputed dates are mentioned:
'   Friday   , May       14th, 1948: Refoundation of Israel
'   Saturday, October 31st, 1517: Luther's theses in Wittenberg
'   Sunday   , April    9th,   30: Resurrection of Christ Jesus [Rusk1974]
'
' [Rusk1974]
' Rusk R. in "Christianity Today", Washington D.C., Vol XIII, Nr. 13,
' (29. 3.1974), pp 4 (720) - 6 (722);
' Without comment translated from the English language into the German
' language by Erhard Kietz, physicist retired
'
' [Zem1987]
' Zemanek H.: "Kalender und Chronologie" (Calendar and Chronology),
' R. Oldenbourg Verlag Munich, Vienna, 4th improved edition, (1987)
'-----
'
' Handling:
' 8/ 5/2001 - 9/ 4/2001:   Norbert Südland, Munich
' 9/ 1/2002 - 9/ 5/2002:   Norbert Südland, Aalen
' Check:
' 9/ 1/2002 - 9/ 5/2002:   Norbert Südland, Aalen
'-----
DIM Century%      'AS INTEGER
DIM Year%         'AS INTEGER
DIM Month%        'AS INTEGER
DIM Day%          'AS INTEGER
DIM Intercalary%  'AS INTEGER
DIM Area%         'AS INTEGER
DIM Rest&         'AS LONG
DIM WDay&         'AS LONG

```

'Preparation:

'-----'

Century% = Ctry%

Year% = y% + INT((m% - 1) / 12)

'years of 12 months only

Month% = ((m% - 1) MOD 12) + 1

'only 12 months are enabled

WHILE Month% < 1

Month% = Month% + 12

'month must own the correct area!

WEND

Day% = d%

'The 0th Century and the 0th Year Do Not Exist:

'-----'

IF Year% < 0 THEN Year% = Year% + 1 'leap-years can be divided by no. 4!

SELECT CASE Century%

CASE IS > 0

Century% = Century% - 1

Year% = 100 * Century% + Year%

CASE IS < 0

Year% = 100 * Century% + Year%

END SELECT

'Orientate the Days Since May 14th, 1948:

'-----'

WDay% = 6

'Friday, May 14th, 1948

WDay% = WDay% + Day% - 14

SELECT CASE Month%

CASE 1

WDay% = WDay% - 120

CASE 2

WDay% = WDay% - 89

CASE 3

WDay% = WDay% - 61

CASE 4

WDay% = WDay% - 30

CASE 6

WDay% = WDay% + 31

CASE 7

WDay% = WDay% + 61

CASE 8

WDay% = WDay% + 92

CASE 9

WDay% = WDay% + 123

CASE 10

WDay% = WDay% + 153

CASE 11

WDay% = WDay% + 184

CASE 12

WDay% = WDay% + 214

END SELECT

'Calculation due to the Egyptian Sun Year:

'-----'

WDay% = WDay% + (Year% - 1948) * 365&

'Julian Intercalary Days Since 45 b.Chr. (The 0th Year Does Not Exist):

'-----'

WDay% = WDay% + INT((FNMAX(-45, Year%) - 1948) / 4)

'Gregorianic Equalization Until the Year 1948 Since October 15th, 1582:

'-----'

WDay% = WDay% - INT(FNMAX(Year%, 1582) / 100 - 19) 'INT(1948 / 100) = 19

WDay% = WDay% + INT(FNMAX(Year%, 1582) / 400 - 4) 'INT(1948 / 400) = 4

IF Year% < 1582 OR (Year% = 1582 AND Month% < 10) THEN

WDay% = WDay% + 10 'bridge the historical gap days

END IF

IF Year% = 1582 AND Month% = 10 AND Day% < 5 THEN

WDay% = WDay% + 10 'bridge the historical gap days

END IF

'Within the Leap-Year Before the Intercalary Day:

'-----'

IF Year% > -45 AND Year% MOD 4 = 0 AND Month% < 3 THEN

WDay% = WDay% - 1 'the intercalary day takes place

'Gregorianic Calendar Reform:

'-----'

IF Year% > 1582 AND Year% MOD 100 = 0 AND Year% MOD 400 <> 0 THEN

WDay% = WDay% + 1 'no intercalary day for normal turn of the century

END IF

END IF

'Adjust the Date Numbers due to the Result:

'-----'

SELECT CASE WDay%

CASE IS <= -727635

'Egyptian Sun Calendar Until February 28th, 45 b.Chr.:

'-----'

' A normal sun year lasts 365 days.

' The difference from 1/ 1/1948 until Friday, 5/14/1948, yields 128 days.

' Until 1948 the Gregorianic Reform omits 13 of the 498 intercalary days.

' The week day number is negative, thus addition is caused by omission.

'-----'

Year% = INT((WDay% + 498 - 13 + 31 + 28 + 31 + 30 + 14 - 6) / 365)

Rest% = (WDay% + 498 - 13 + 128) - Year% * 365

Year% = Year% + 1948

Intercalary% = 0

CASE -727634 TO -133520

'Julian Calendar From February 29th, 45 b.Chr. Until 10/ 4/1582 p.Chr.:

'-----'

' A 4 Year Periode lasts 1461 days.

' A normal leap-year lasts 365 days.

' The last block of a periode can own an additional duration.

' The difference from 1/1/1945 until Friday, 5/14/1948 yields 1223 days.

' This bridge to the Gregorianic Reform omits 13 days.

'-----'

'Attach the Year Quartet:

'-----'

Area% = FNMIN(INT((WDay% + 1223 - 13) / 1461), 3)

Rest% = (WDay% + 1223 - 13) - Area% * 1461

Year% = 1945 + 4 * Area%

Intercalary% = 0

'The Last Year in the Year Quartet Might Last More Time:

'-----'

Area% = FNMIN(INT(Rest% / 365), 3)

Rest% = Rest% - Area% * 365

Year% = Year% + Area%

IF Area% = 3 THEN

Intercalary% = Intercalary% + 1 'Mark the Leap-Year

END IF

CASE IS >= -133519

```
'Gregorianic Calendar Reform Since 10/15/1582 p.Chr.:
'-----
' A 400 year periode lasts 146097 days ([Zem1987], page 29).
' A 100 year periode lasts 36524 days.
' A 4 year periode lasts 1461 days.
' A normal sun year lasts 365 days.
' Each last block of a periode might own an additional duration.
' The difference from 1/1/1601 until Friday 5/14/1948 yields 126866 days.
'-----
```

```
'Attach the Century Quartet:
'-----
```

```
Area% = INT((WDay& + 126866) / 146097)
Rest& = (WDay& + 126866) - Area% * 146097
Year% = 1601 + 400 * Area%
Intercalary% = 0
```

```
'The Last Century of the Quartet Needs More Time:
'-----
```

```
Area% = FNMIN(INT(Rest& / 36524), 3)
Rest& = Rest& - Area% * 36524
Year% = Year% + 100 * Area%
IF Area% = 3 THEN
    Intercalary% = 1                'Mark the Leap-Century
END IF
```

```
'Each Century Owns 25 Year Quartets:
'-----
```

```
Area% = FNMIN(INT((Rest&) / 1461), 24)
Rest& = Rest& - Area% * 1461&
Year% = Year% + 4 * Area%
IF Area% = 24 AND Rest& >= 3 * 365 THEN
    Intercalary% = Intercalary% - 1    'Usually no Leap-Year at Century End
ELSE
    Intercalary% = 0
END IF
```

```
'The last Year of the Year Quartet Might Last More Time:
'-----
```

```
Area% = FNMIN(INT(Rest& / 365), 3)
Rest& = Rest& - Area% * 365&
Year% = Year% + Area%
IF Area% = 3 THEN
    Intercalary% = Intercalary% + 1    'Mark the Leap-Year
END IF
```

END SELECT

```
'Attach the Date Numbers:
'-----
```

```
IF Year% < 1 THEN Year% = Year% - 1
SELECT CASE Rest&
CASE 0 TO 30
    Month% = 1
    Day% = Rest& + 1
CASE 31 TO 58 + Intercalary%
```

```

    Month% = 2
    Day% = Rest& - 30
CASE 59 + Intercalary% TO 89 + Intercalary%
    Month% = 3
    Day% = Rest& - 58 - Intercalary%
CASE 90 + Intercalary% TO 119 + Intercalary%
    Month% = 4
    Day% = Rest& - 89 - Intercalary%
CASE 120 + Intercalary% TO 150 + Intercalary%
    Month% = 5
    Day% = Rest& - 119 - Intercalary%
CASE 151 + Intercalary% TO 180 + Intercalary%
    Month% = 6
    Day% = Rest& - 150 - Intercalary%
CASE 181 + Intercalary% TO 211 + Intercalary%
    Month% = 7
    Day% = Rest& - 180 - Intercalary%
CASE 212 + Intercalary% TO 242 + Intercalary%
    Month% = 8
    Day% = Rest& - 211 - Intercalary%
CASE 243 + Intercalary% TO 272 + Intercalary%
    Month% = 9
    Day% = Rest& - 242 - Intercalary%
CASE 273 + Intercalary% TO 303 + Intercalary%
    Month% = 10
    Day% = Rest& - 272 - Intercalary%
CASE 304 + Intercalary% TO 333 + Intercalary%
    Month% = 11
    Day% = Rest& - 303 - Intercalary%
CASE 334 + Intercalary% TO 364 + Intercalary%
    Month% = 12
    Day% = Rest& - 333 - Intercalary%
CASE ELSE
    Pause                'Programming Error Within the Week Day Calculation!
    PRINT Rest&
    SYSTEM
END SELECT

Ctry% = Century%
IF Ctry% >= 0 THEN Ctry% = Ctry% + 1 'Century value is superior.
y% = Year%                        'Year Value is accepted completely.
m% = Month%
d% = Day%

'Concrete Number of the Week Day:
'-----'
WDay& = WDay& MOD 7&
WHILE WDay& < 1
    WDay& = WDay& + 7&
WEND

WeekDay% = WDay&
END FUNCTION 'WeekDay% _____'

'=====
SUB WriteTo (File%, DataSetLength%, Position&, Data$)
'=====
' Fits `Data$` to `DataSetLength%` and writes them to the corresponding

```

```
' `Position&` of a binary `File%` owning a constant `DataSetLength%`.
' The 0th file record is used for documentation.
'
' Handling:
' 1/21/2003: Norbert Suedland
' Translation:
' 11/24/2007: Norbert Suedland
'-----'
DIM Buffer$ 'AS STRING

Buffer$ = LEFT$(Data$, DataSetLength%)
Buffer$ = SPACE$(DataSetLength% - LEN(Buffer$)) + Buffer$
PUT #File%, Position& * DataSetLength% + 1, Buffer$
END SUB 'WriteTo _____'
```