
@ECHO OFF 

REM =========================================================================== 

REM  HISTORIC.BAT        1.007                                       4/26/2025 

REM =========================================================================== 

REM  Starting Programm for HISTORIC in QBASIC 

REM =========================================================================== 

REM  Feature: 

REM  Jump labels are considered by MS-DOS 3.30 at their first 8 letters only! 

REM =========================================================================== 

REM  Handling: 11/29/1994 -  4/26/2025 Norbert Suedland 

REM =========================================================================== 

 

REM A Temporary Directory is Set and QBASIC is Installed: 

REM ----------------------------------------------------- 

REM There is also the possibility to set HISTORICTEMP directly before! 

REM ----------------------------------------------------------------- 

IF NOT x%DOSDRV%x==xx SET SYSTEMDRIVE=%DOSDRV% 

IF x%SYSTEMDRIVE%x==xx SET SYSTEMDRIVE=C: 

IF NOT x%TEMP%x==xx GOTO 2_Next 

IF NOT x%TMP%x==xx GOTO 1_Next 

MKDIR %SYSTEMDRIVE%\TEMP 

SET TEMP=%SYSTEMDRIVE%\TEMP 

GOTO 2_Next 

:1_Next 

SET TEMP=%TMP% 

:2_Next 

SET QBASICTEMP=%TEMP% 

IF NOT EXIST INSTALL.BAT GOTO 1_Error 

CALL INSTALL.BAT 

 

REM The Program is Fixed to the ASCII CODEPAGE No. 437: 

REM --------------------------------------------------- 

IF x%WINDIR%x==xx GOTO 1_DosPath 

IF EXIST %WINDIR%\SYSTEM32\EGA.CPI GOTO WindowsNT 

IF EXIST %WINDIR%\COMMAND\EGA.CPI GOTO Windows95 

:1_DosPath 

IF EXIST %SYSTEMDRIVE%\DOS\EGA.CPI GOTO Dos 

IF EXIST %DOSDIR%\CPI\EGA.CPX GOTO FreeDos 

KEYB US 437 

IF NOT EXIST Z:\COMMAND.COM GOTO Emergency 

IF NOT EXIST Z:\MOUNT.COM GOTO Emergency 

IF NOT EXIST Z:\INTRO.COM GOTO Emergency 

Goto Start 

:Emergency 

ECHO File EGA.CPI not found. 

ECHO Therefore now the CODEPAGE 437 is assumed to be valid. 

PAUSE 

GOTO Start 

:Dos 

MODE CON CODEPAGE PREPARE=((437) %SYSTEMDRIVE%\DOS\EGA.CPI) 

KEYB US,437,%SYSTEMDRIVE%\DOS\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=437 

GOTO Start 

:FreeDos 

SET LANG=EN 

SET DIRCMD=/Y 

LH DISPLAY CON=(EGA,,1) 

MODE CON CP PREP=((437) %DOSDIR%\CPI\EGA.CPX) 

MODE CON CP SEL=437 

MODE CON CP REFRESH 

MKEYB US 

GOTO Start 

:Windows95 

MODE CON CODEPAGE PREPARE=((437) %WINDIR%\COMMAND\EGA.CPI) 

IF EXIST %WINDIR%\COMMAND\KEYB.COM KEYB US,437,%WINDIR%\COMMAND\KEYBOARD.SYS 

IF NOT EXIST %WINDIR%\COMMAND\KEYB.COM KEYBOARD 

US,437,%WINDIR%\COMMAND\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=437 

GOTO Start 

:WindowsNT 

MODE CON CODEPAGE PREPARE=((437) %WINDIR%\SYSTEM32\EGA.CPI) 

REM KEYB US,437,%WINDIR%\SYSTEM32\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=437 

CHCP 437 

 

REM The Proper Start of the Program: 

REM -------------------------------- 

:Start 

IF NOT EXIST HISTORIC.BAS GOTO 2_Error 

ECHO Trial to load the programm ″QBASIC.EXE″: 

QBASIC.EXE /H /RUN HISTORIC.BAS 

 

REM Eventually Printing of Files: 

REM ----------------------------- 

IF x%HISTORICTEMP%x==xx GOTO 3_Next 

IF NOT EXIST %HISTORICTEMP%\PRINT.YES GOTO COMPARE 

ECHO Please prepare the printer to work. 

PAUSE 

ECHO Print small and double: It's your turn! 

@ECHO ON 

MY_PRINT %HISTORICTEMP%\*.PR* S D 

@ECHO OFF 

DEL %HISTORICTEMP%\PRINT.YES 

PAUSE 

GOTO Start 

 

:3_Next 

IF x%QBASICTEMP%x==xx GOTO 4_Next 

IF NOT EXIST %QBASICTEMP%\HISTORIC.TMP\PRINT.YES GOTO COMPARE 

ECHO Please prepare the printer to work. 

PAUSE 

ECHO Print small and double: It's your turn! 

@ECHO ON 

MY_PRINT %QBASICTEMP%\HISTORIC.TMP\*.PR* S D 

@ECHO OFF 

DEL %QBASICTEMP%\HISTORIC.TMP\PRINT.YES 

PAUSE 

GOTO Start 

 

:4_Next 

IF NOT EXIST HISTORIC.TMP\PRINT.YES GOTO Compare 

ECHO Please prepare the printer to work. 

PAUSE 

ECHO Print small and double: It's your turn! 

@ECHO ON 

MY_PRINT HISTORIC.TMP\*.PR* S D 

@ECHO OFF 

DEL HISTORIC.TMP\PRINT.YES 

PAUSE 

GOTO Start 

 

REM  Error Handling: 

REM ----------------- 

:1_Error 

ECHO The program ″INSTALL.BAT″ is missing. 

PAUSE 

GOTO End 

:2_Error 

ECHO The program ″HISTORIC.BAS″ is missing. 

PAUSE 

 

REM Compare Finally the Results: 

REM ---------------------------- 

:COMPARE 

QBASIC.EXE /H /RUN COMPARE.BAS 

 

 

 

REM Reset CODEPAGE to Windows Default or Another Value: 

REM --------------------------------------------------- 

:End 

IF x%WINDIR%x==xx GOTO 2_DosPath 

IF EXIST %WINDIR%\SYSTEM32\EGA.CPI GOTO EndWindowsNT 

IF EXIST %WINDIR%\COMMAND\EGA.CPI GOTO EndWindows95 

:2_DosPath 

IF EXIST %SYSTEMDRIVE%\DOS\EGA.CPI GOTO EndDos 

IF EXIST %DOSDIR%\CPI\EGA.CPX GOTO EndFreeDos 

IF NOT EXIST Z:\COMMAND.COM GOTO Finish 

IF NOT EXIST Z:\MOUNT.COM GOTO Finish 

IF NOT EXIST Z:\INTRO.COM GOTO Finish 

ECHO To end the DOS Window, please put in: EXIT 

GOTO Finish 

:EndDos 

MODE CON CODEPAGE PREPARE=((850) %SYSTEMDRIVE%\DOS\EGA.CPI) 

KEYB US,850,%SYSTEMDRIVE%\DOS\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=850 

GOTO Finish 

:EndFreeDos 

MODE CON CP PREP=((858) %DOSDIR%\CPI\EGA.CPX) 

MODE CON CP SEL=858 

MODE CON CP REFRESH 

SET DIRCMD=\P \ONG \Y 

GOTO Finish 

:EndWindows95 

MODE CON CODEPAGE PREPARE=((850) %WINDIR%\COMMAND\EGA.CPI) 

IF EXIST %WINDIR%\COMMAND\KEYB.COM KEYB US,850,%WINDIR%\COMMAND\KEYBOARD.SYS 

IF NOT EXIST %WINDIR%\COMMAND\KEYB.COM KEYBOARD 

US,850,%WINDIR%\COMMAND\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=850 

GOTO Finish 

:EndWindowsNT 

MODE CON CODEPAGE PREPARE=((850) %WINDIR%\SYSTEM32\EGA.CPI) 

REM KEYB US,850,%WINDIR%\SYSTEM32\KEYBOARD.SYS 

MODE CON CODEPAGE SELECT=850 

CHCP 850 

 

REM A PAUSE at the End Guarantees the Readability of All Error Messages: 

REM -------------------------------------------------------------------- 

:Finish 

PAUSE 

 

REM =========================================================================== 

REM  End of HISTORIC.BAT 

REM =========================================================================== 


